

お客様各位

カタログ等資料中の旧社名の扱いについて

2010年4月1日を以ってNECエレクトロニクス株式会社及び株式会社ルネサステクノロジが合併し、両社の全ての事業が当社に承継されております。従いまして、本資料中には旧社名での表記が残っておりますが、当社の資料として有効ですので、ご理解の程宜しくお願ひ申し上げます。

ルネサスエレクトロニクス ホームページ (<http://www.renesas.com>)

2010年4月1日
ルネサスエレクトロニクス株式会社

【発行】ルネサスエレクトロニクス株式会社 (<http://www.renesas.com>)

【問い合わせ先】<http://japan.renesas.com/inquiry>

ご注意書き

1. 本資料に記載されている内容は本資料発行時点のものであり、予告なく変更することがあります。当社製品のご購入およびご使用にあたりましては、事前に当社営業窓口で最新の情報をご確認いただきますとともに、当社ホームページなどを通じて公開される情報に常にご注意ください。
2. 本資料に記載された当社製品および技術情報の使用に関連し発生した第三者の特許権、著作権その他の知的財産権の侵害等に関し、当社は、一切その責任を負いません。当社は、本資料に基づき当社または第三者の特許権、著作権その他の知的財産権を何ら許諾するものではありません。
3. 当社製品を改造、改変、複製等しないでください。
4. 本資料に記載された回路、ソフトウェアおよびこれらに関する情報は、半導体製品の動作例、応用例を説明するものです。お客様の機器の設計において、回路、ソフトウェアおよびこれらに関する情報を使用する場合には、お客様の責任において行ってください。これらの使用に起因しお客様または第三者に生じた損害に関し、当社は、一切その責任を負いません。
5. 輸出に際しては、「外国為替及び外国貿易法」その他輸出関連法令を遵守し、かかる法令の定めるところにより必要な手続を行ってください。本資料に記載されている当社製品および技術を大量破壊兵器の開発等の目的、軍事利用の目的その他軍事用途の目的で使用しないでください。また、当社製品および技術を国内外の法令および規則により製造・使用・販売を禁止されている機器に使用することができません。
6. 本資料に記載されている情報は、正確を期すため慎重に作成したものですが、誤りがないことを保証するものではありません。万一、本資料に記載されている情報の誤りに起因する損害がお客様に生じた場合においても、当社は、一切その責任を負いません。
7. 当社は、当社製品の品質水準を「標準水準」、「高品質水準」および「特定水準」に分類しております。また、各品質水準は、以下に示す用途に製品が使われることを意図しておりますので、当社製品の品質水準をご確認ください。お客様は、当社の文書による事前の承諾を得ることなく、「特定水準」に分類された用途に当社製品を使用することができません。また、お客様は、当社の文書による事前の承諾を得ることなく、意図されていない用途に当社製品を使用することができません。当社の文書による事前の承諾を得ることなく、「特定水準」に分類された用途または意図されていない用途に当社製品を使用したことによりお客様または第三者に生じた損害等に関し、当社は、一切その責任を負いません。なお、当社製品のデータ・シート、データ・ブック等の資料で特に品質水準の表示がない場合は、標準水準製品であることを表します。

標準水準： コンピュータ、OA機器、通信機器、計測機器、AV機器、家電、工作機械、パソコン機器、産業用ロボット

高品質水準： 輸送機器（自動車、電車、船舶等）、交通用信号機器、防災・防犯装置、各種安全装置、生命維持を目的として設計されていない医療機器（厚生労働省定義の管理医療機器に相当）

特定水準： 航空機器、航空宇宙機器、海底中継機器、原子力制御システム、生命維持のための医療機器（生命維持装置、人体に埋め込み使用するもの、治療行為（患部切り出し等）を行うもの、その他直接人命に影響を与えるもの）（厚生労働省定義の高度管理医療機器に相当）またはシステム等

8. 本資料に記載された当社製品のご使用につき、特に、最大定格、動作電源電圧範囲、放熱特性、実装条件その他諸条件につきましては、当社保証範囲内でご使用ください。当社保証範囲を超えて当社製品をご使用された場合の故障および事故につきましては、当社は、一切その責任を負いません。
9. 当社は、当社製品の品質および信頼性の向上に努めておりますが、半導体製品はある確率で故障が発生したり、使用条件によっては誤動作したりする場合があります。また、当社製品は耐放射線設計については行っておりません。当社製品の故障または誤動作が生じた場合も、人身事故、火災事故、社会的損害などを生じさせないようお客様の責任において冗長設計、延焼対策設計、誤動作防止設計等の安全設計およびエーペンギング処理等、機器またはシステムとしての出荷保証をお願いいたします。特に、マイコンソフトウェアは、単独での検証は困難なため、お客様が製造された最終の機器・システムとしての安全検証をお願いいたします。
10. 当社製品の環境適合性等、詳細につきましては製品個別に必ず当社営業窓口までお問合せください。ご使用に際しては、特定の物質の含有・使用を規制する RoHS 指令等、適用される環境関連法令を十分調査のうえ、かかる法令に適合するようご使用ください。お客様がかかる法令を遵守しないことにより生じた損害に関して、当社は、一切その責任を負いません。
11. 本資料の全部または一部を当社の文書による事前の承諾を得ることなく転載または複製することを固くお断りいたします。
12. 本資料に関する詳細についてのお問い合わせその他お気付きの点等がございましたら当社営業窓口までご照会ください。

注 1. 本資料において使用されている「当社」とは、ルネサスエレクトロニクス株式会社およびルネサスエレクトロニクス株式会社がその総株主の議決権の過半数を直接または間接に保有する会社をいいます。

注 2. 本資料において使用されている「当社製品」とは、注 1 において定義された当社の開発、製造製品をいいます。

μPD750008サブシリーズ

4ビット・シングルチップ・マイクロコンピュータ

μPD750004
μPD750006
μPD750008
μPD75P0016

第1章 概 説

1

第2章 システム・クロック切り替え機能の応用

2

第3章 ベーシック・インターバル・タイマの応用

3

第4章 タイマ／イベント・カウンタの応用

4

第5章 時計用タイマの応用

5

第6章 クロック出力回路の応用

6

第7章 ビット・シーケンシャル・バッファの応用

7

第8章 シリアル・インターフェースの応用

8

第9章 キー入力サブルーチン

9

第10章 各種サブルーチン

10

第11章 このアプリケーション・プログラムの応用例

11

(× も)

本製品が外国為替および外国貿易管理法の規定による戦略物資等（または役務）に該当するか否かは、ユーザ（仕様を決定した者）が判定してください。

本資料に掲載の応用回路および回路定数は、例示的に示したものであり、量産設計を対象とするものではありません。

- 本資料の内容は、後日変更する場合があります。
- 文書による当社の承諾なしに本資料の転載複製を禁じます。
- 本資料に記載された製品の使用もしくは本資料に記載の情報の使用に際して、当社は当社もしくは第三者の知的所有権その他の権利に対する保証または実施権の許諾を行うものではありません。上記使用に起因する第三者所有の権利にかかる問題が発生した場合、当社はその責を負うものではありませんのでご了承ください。
- 当社は品質、信頼性の向上に努めていますが、半導体製品はある確率で故障が発生します。当社半導体製品の故障により結果として、人身事故、火災事故、社会的な損害等を生じさせない冗長設計、延焼対策設計、誤動作防止設計等安全設計に十分ご注意願います。
- 当社は、当社製品の品質水準を「標準水準」、「特別水準」およびお客様に品質保証プログラムを指定して頂く「特定水準」に分類しております。また、各品質水準は以下に示す用途に製品が使われることを意図しておりますので、当社製品の品質水準をご確認の上ご使用願います。
標準水準：コンピュータ、OA機器、通信機器、計測機器、AV機器、家電、工作機械、パソコン機器、産業用ロボット
特別水準：輸送機器（自動車、列車、船舶等）、交通用信号機器、防災／防犯装置、各種安全装置、生命維持を直接の目的としない医療機器
特定水準：航空機器、航空宇宙機器、海底中継機器、原子力制御システム、生命維持のための医療機器、生命維持のための装置またはシステム等
- 当社製品のデータ・シート／データ・ブック等の資料で、特に品質水準の表示がない場合は標準水準製品であることを表します。当社製品を上記の「標準水準」の用途以外でご使用をお考えのお客様は、必ず事前に当社販売窓口までご相談頂きますようお願い致します。
- この製品は耐放射線設計をしておりません。

巻末にアンケート・コーナを設けております。このドキュメントに対するご意見をお気軽にお寄せください。

はじめに

対象者 このアプリケーション・ノートは μ PD750008サブシリーズの機能を理解し、それを用いたアプリケーション・プログラムを設計するユーザのエンジニアを対象とします。

μ PD750008サブシリーズとは、 μ PD750004, 750006, 750008, 75P0016の総称です。

目的 このアプリケーション・ノートは、 μ PD750008サブシリーズの持つ機能を、応用プログラム例を用いてユーザに理解していただくことを目的としています。

読み方 このアプリケーション・ノートをご利用なさるお客様には、電気、論理回路、およびマイクロコンピュータに関する一般知識を必要とします。

また、特に機能面で違いがない限りは、 μ PD750008を代表品種として記述しています。

μ PD750004, 750006, 75P0016のマニュアルとしてお使いの場合は、 μ PD750008をそれぞれの品種に読み替えてご利用ください。

凡例 データ表記の重み : 左が上位桁、右が下位桁

アクティブ・ロウの表記 : \overline{XXX} (端子、信号名称に上線)

注 : 本文中につけた注の説明

注意 : 気をつけて読んでいただきたい内容

備考 : 本文の補足説明

数の表記 : 2進数・・・ $\times \times \times$ または $\times \times \times B$

10進数・・・ $\times \times \times$

16進数・・・ $\times \times \times H$

関連資料

デバイス関連資料

資料 製品	パンフレット	データ・シート	ユーザーズ・ マニュアル	インストラクション 活用表	アプリケーション・ ノート
μ PD750004	IF-6367	U10738J	IEU-884	IEM-5593	U10452J
μ PD750006					(この資料)
μ PD750008					
μ PD75P0016		U10328J			

(× 穂)

目 次

第1章 概 説 … 1

- 1.1 μ PD75008と μ PD750008との違い … 1
- 1.2 Mk I モードとMk II モードの切り替え … 4
 - 1.2.1 Mk I モードとMk II モードの使用法 … 4
 - 1.2.2 レジスタ・バンクの使用法 … 6
- 1.3 アプリケーション・プログラムの説明 … 11

第2章 システム・クロック切り替え機能の応用 … 13

- 2.1 RESET後のPCCの切り替え … 15
- 2.2 商用電源の停電検出時のシステム・クロックの切り替え … 17
 - 2.2.1 パワーオフ処理 … 18
 - 2.2.2 パワーオン処理 … 18
 - 2.2.3 パワーオン／オフ処理の応用 … 19

第3章 ベーシック・インターバル・タイマの応用 … 23

- 3.1 基準時間発生 … 23
- 3.2 ウォッチドッグ・タイマの応用 … 25
- 3.3 リモコン信号受信応用 … 27

第4章 タイマ／イベント・カウンタの応用 … 43

- 4.1 インターバル・タイマの設定 … 43
- 4.2 PTO端子への出力例 … 46
 - 4.2.1 メロディ出力 … 46
 - 4.2.2 イベント・パルス分周出力 … 57
- 4.3 ワンショット・パルス出力 … 59

第5章 時計用タイマの応用 … 65

- 5.1 時計プログラム … 65

第6章 クロック出力回路の応用 … 71

- 6.1 PCLクロック出力 … 71

第7章 ビット・シーケンシャル・バッファの応用 … 73

- 7.1 高速シリアル・データ転送 … 74

第8章 シリアル・インターフェースの応用 … 87

 8.1 SBIモードによる応用例 … 87

第9章 キー入力サブルーチン … 99

第10章 各種サブルーチン … 107

 10.1 データ転送 … 107
 10.2 データ比較 … 108
 10.3 10進加算 … 109
 10.4 10進減算 … 110
 10.5 2進10進変換 … 112
 10.6 8ビット乗算 … 115
 10.7 16ビット乗算 … 117
 10.8 2進除算 (16ビット÷8ビット) … 119
 10.9 2進除算 (16ビット÷16ビット) … 122

第11章 このアプリケーション・プログラムの応用例 … 125

 11.1 プログラム概要 … 125
 11.2 システム構成 … 127
 11.3 ポート割り付け … 128
 11.4 状態遷移表 … 130
 11.5 キーごとの機能説明 … 131
 11.5.1 MODEキー … 131
 11.5.2 UP/DOWNキー … 132
 11.5.3 音階キー … 133
 11.5.4 ENDキー … 134
 11.5.5 その他のキー … 134
 11.6 表示説明 … 135
 11.6.1 LED表示 … 135
 11.6.2 LCD表示 … 137
 11.7 ハードウェア構成 … 138
 11.8 アプリケーション・プログラム … 139

図の目次 (1/2)

図番号	タイトル, ページ
1-1	スタック・バンク選択レジスタのフォーマット … 5
1-2	レジスタ・バンクの使い分け例 … 7
1-3	汎用レジスタの構成（4ビット処理の場合） … 9
1-4	汎用レジスタの構成（8ビット処理の場合） … 10
2-1	プロセッサ・クロック・コントロール・レジスタのフォーマット … 13
2-2	システム・クロック・コントロール・レジスタのフォーマット … 14
2-3	電源電圧（V _{DD} ）に対する最小命令実行時間（t _{CY} ） … 14
2-4	RESET 後のCPUクロック切り替え … 16
2-5	商用電源オン／オフ時のシステム・クロックの切り替え … 17
2-6	INT4割り込み処理のアルゴリズム … 17
3-1	ベーシック・インターバル・タイマ・モード・レジスタのフォーマット … 23
3-2	ウォッチドッグ・タイマ許可フラグのフォーマット … 25
3-3	リモコン信号受信回路例 … 27
3-4	リモコン送信用IC出力信号 … 28
3-5	受信プリアンプの出力波形 … 29
4-1	PTO端子の使用例 … 46
4-2	P20/PTO0端子からの出力信号の概念図 … 47
4-3	データ・フォーマット … 48
4-4	イベント・パルス分周出力 … 57
4-5	ワンショット・パルス出力 … 59
4-6	ワンショット・パルス出力のタイミング（1） … 59
4-7	ワンショット・パルス出力のタイミング（2） … 60
4-8	ワンショット・パルス出力のタイミング（3） … 60
4-9	プログラム上のRAMのレイアウト … 61
6-1	クロック出力モード・レジスタのフォーマット … 71
6-2	クロック出力波形 … 72
7-1	ビット・シーケンシャル・バッファの間接アドレッシング … 73
7-2	マルチプロセッサ・システム構成図（高速シリアル・データ転送） … 74
7-3	プログラム上のRAMのレイアウト … 76
8-1	シリアル・バスの構成例 … 87

図の目次 (2/2)

図番号	タイトル, ページ
8-2 アドレス, コマンド, データのフォーマット	… 88
8-3 アクノリッジ信号によるエラーの判断	… 89
9-1 キー・マトリクスの構成	… 99
11-1 システム構成図	… 127

表の目次

表番号	タイトル, ページ
1-1	μ PD750008サブシリーズの各製品間の違い … 1
1-2	μ PD75008と μ PD750008との違い … 2
1-3	Mk I モードとMk II モードの違い … 4
1-4	RBE, RBSと選択されるレジスタ・バンク … 6
1-5	通常ルーチンと割り込みルーチンでのレジスタ・バンクの使い分けの例 … 6
4-1	分解能と最長設定時間 ($f_x = 4.194304 \text{ MHz}$) … 44
4-2	各音階に対してモジュロ・レジスタに設定すべき値, およびそれによって得られる周波数の各音階周波数に対する誤差 … 47
4-3	音階データと音階, 表示, およびPTO0端子出力の関係 … 49
4-4	音長データと音階出力の関係 … 49
11-1	μ PD750008のポート割り付け … 128
11-2	状態遷移表 … 130

(x - 7)

第1章 概 説

1

μ PD750008サブシリーズは、豊富な製品展開を誇る75Xシリーズの後継品種、75XLシリーズの4ビット・シングルチップ・マイクロコンピュータです。

μ PD750008サブシリーズの各製品間の違いを表1-1に示します。

表1-1 μ PD750008サブシリーズの各製品間の違い

品名	プログラム・メモリ(ROM)	ROMの構成	プログラム・カウンタ
μ PD750004	4096バイト	マスクROM	12ビット
μ PD750006	6144バイト		13ビット
μ PD750008	8192バイト		
μ PD75P0016	16384バイト	ワン・タイムPROM	14ビット

1.1 μ PD75008と μ PD750008との違い

μ PD750008は、従来製品である μ PD75008(75Xシリーズ)の機能やインストラクションを継承しており、置き換えが容易です。

表1-2に μ PD75008と μ PD750008との違いを示します。

表1-2 μ PD75008と μ PD750008との違い (1/2)

項 目		μ PD75008	μ PD750008
プログラム・メモリ		0000H-1F7FH 8064×8ビット	0000H-1FFFFH 8192×8ビット
データ・メモリ		000H-1FFFH (512×4ビット)	
CPU		75X Standard	75XL
発振安定ウエイト時間		31.3 msに固定	2 ¹⁵ /fx, 2 ¹⁷ /fx ^注 (マスク・オプションで選択可能)
命令実行時間	メイン・システム・クロック選択時	0.95, 1.91, 15.3 μ s (4.19 MHz動作のみ)	• 0.95, 1.91, 3.81, 15.3 μ s (4.19 MHz動作時) • 0.67, 1.33, 2.67, 10.7 μ s (6.0 MHz動作時)
	サブシステム・クロック選択時	122 μ s (32.768 kHz動作時)	
スタック	SBSレジスター	なし	あり • SBS.3 = 1 : Mk I モード選択 • SBS.3 = 0 : Mk II モード選択
	スタック・エリア	000H-0FFFH	n00H-nFFH (n = 0, 1)
	サブルーチン・コール命令の スタック動作	2 バイト・スタック	Mk I モード時：2 バイト・スタック Mk II モード時：3 バイト・スタック
命令	BRA laddr1動作 CALLA laddr1動作	使用不可	Mk I モード時：使用不可 Mk II モード時：使用可能
	MOVT XA, @BCDE MOVT XA, @BCXA BR BCDE BR BCXA		使用可能
	CALL laddr	3 マシン・サイクル	Mk I モード時：3 マシン・サイクル Mk II モード時：4 マシン・サイクル
	CALLF lfaddr	2 マシン・サイクル	Mk I モード時：2 マシン・サイクル Mk II モード時：3 マシン・サイクル
	タイマ	3 チャネル • ベーシック・インターバル・タイマ • タイマ／イベント・カウンタ • 時計用タイマ	4 チャネル • ベーシック・インターバル・タイマ／ウォッチドッグ・タイマ • タイマ／イベント・カウンタ • タイマ・カウンタ • 時計用タイマ
	クロック出力 (PCL)	• Φ , 524, 262, 65.5 kHz (4.19 MHz動作時)	• Φ , 524, 262, 65.5 kHz (4.19 MHz動作時) • Φ , 750, 375, 93.7 kHz (6.0 MHz動作時)

注 2¹⁵/fxは、6.0 MHz時：5.46 ms, 4.19 MHz時：7.81 ms。

2¹⁷/fxは、6.0 MHz時：21.8 ms, 4.19 MHz時：31.3 ms。

表1-2 μ PD75008と μ PD750008との違い (2/2)

項目	μ PD75008	μ PD750008
ブザー出力 (BUZ)	2 kHz	<ul style="list-style-type: none"> • 2.4, 32 kHz (4.19 MHz動作時) • 2.86, 5.72, 45.8 kHz (6.0 MHz動作時)
シリアル・インターフェース	3種類のモードに対応可能 <ul style="list-style-type: none"> • 3線式シリアルI/Oモード…MSB/LSB先頭切り替え可能 • 2線式シリアルI/Oモード • SBIモード 	
SOSレジスタ	フィードバック抵抗カット・フラグ (SOS.0)	マスク・オプションによりフィードバック抵抗を内蔵可能
	サブ発振器電流カット・フラグ (SOS.1)	なし
レジスタ・バンク選択レジスタ (RBS)	なし	あり
INT0によるスタンバイ解除	不可	可能
ベクタ割り込み	外部：3本、内部：3本	外部：3本、内部：4本
プロセッサ・クロック・コントロール・レジスタ (PCC)	PCC = 0, 2, 3を使用可能	PCC = 0-3を使用可能
電源電圧	V _{DD} = 2.7~6.0 V	V _{DD} = 2.2~5.5 V
パッケージ	<ul style="list-style-type: none"> • 42ピン・プラスチック・シュリンクDIP (600 mil) • 44ピン・プラスチックQFP (\square10 mm) 	

1.2 Mk I モードとMk II モードの切り替え

1.2.1 Mk I モードとMk II モードの使用法

μ PD750008サブシリーズのCPUはMk I モードとMk II モードの2つのモードを持ち、どちらを使用するかの選択ができます。モードの切り替え操作は、スタック・バンク選択レジスタ（SBS）のビット3で行います。

表1-3にMk I モードとMk II モードの違いを、また図1-1にスタック・バンク選択レジスタのフォーマットを示します。

- ・Mk I モード： μ PD75008サブシリーズと上位互換性があります。

ROM容量が16 Kバイトまでの75XL CPUで使用できます。

- ・Mk II モード： μ PD75008サブシリーズとの互換性がありません。

ROM容量が16 Kバイト以上の製品も含め、75XL CPU全部で使用できます。

表1-3 Mk I モードとMk II モードの違い


	Mk I モード	Mk II モード
サブルーチン命令の スタック・バイト数	2バイト	3バイト
BRA !addr1命令	不定動作	正常動作
CALLA !addr1命令		
CALL !addr命令	3マシン・サイクル	4マシン・サイクル
CALLF !faddr命令	2マシン・サイクル	3マシン・サイクル

備考 Mk II モードは、プログラム・メモリが24 Kバイト以上の75Xシリーズや75XLシリーズとのソフトウェア上の互換性を保つためのものです。

したがって、ROM効率やスピードを重視する場合はMk I モードを使用してください。

スタック・バンク選択レジスタは、4ビット・メモリ操作命令により設定します。Mk I モードを使用する場合は、プログラムの初期で必ずスタック・バンク選択レジスタを $10XXB$ ^注にイニシャライズしてください。またMk II モードを使用する場合は、必ず $00XXB$ ^注にイニシャライズしてください。

図1-1 スタック・バンク選択レジスタのフォーマット



注 $\times \times$ には希望の値を設定してください。

注意 SBS.3はRESET信号発生後“1”になるので、CPUはMk I モードで動作します。Mk II モードの命令を使用する場合は、SBS.3を“0”にし、Mk II モードに設定してから使用してください。

1.2.2 レジスタ・バンクの使用法

μ PD750008サブシリーズは、X,A,B,C,D,E,H,Lの8つの汎用レジスタを1バンクとして、4つのレジスタ・バンクを内蔵しています。この汎用レジスタ・エリアはデータ・メモリのメモリ・バンク0の00H-1FH番地にマッピングされています（図1-3参照）。この汎用レジスタのバンクを指定するためにレジスタ・バンク許可フラグ（RBE）とレジスタ・バンク選択レジスタ（RBS）が内蔵されています。RBSはレジスタ・バンクを選択するためのレジスタで、RBEはRBSで選択されたレジスタ・バンクを有効とするか否かを決定するフラグです。命令実行の際に有効となるレジスタ・バンク（RB）は、次のようにになります。

$$RB = RBE \cdot RBS$$

表1-4 RBE, RBSと選択されるレジスタ・バンク

RBE	RBS				レジスタ・バンク
	3	2	1	0	
0	0	0	X	X	バンク0に固定
1	0	0	0	0	バンク0を選択
			0	1	バンク1 //
			1	0	バンク2 //
			1	1	バンク3 //

↑
↑
0に固定

備考 X : don't care


RBEは、サブルーチン処理時に自動的に退避／復帰されますので、サブルーチン処理中に自由に設定することができます。また、割り込み処理時は、自動的に退避／復帰されるとともに、割り込みベクタ・テーブルの設定によって、割り込み処理開始と同時に、割り込み処理中のRBEを設定することができます。したがって、表1-5に示すように、通常処理と割り込み処理で、レジスタ・バンクを使い分ければ、一重割り込みでは、汎用レジスタの退避／復帰は不要、二重割り込みでは、RBSの退避／復帰のみとなり、割り込み処理の高速化が図れます。

表1-5 通常ルーチンと割り込みルーチンでのレジスタ・バンクの使い分けの例

通常の処理	RBE=1とし、レジスタ・バンク2, 3を使用する。
1重割り込み処理	RBE=0とし、レジスタ・バンク0を使用する。
2重割り込み処理	RBE=1とし、レジスタ・バンク1を使用する。 (このとき、RBSの退避／復帰が必要)
3重以上の割り込み処理	PUSH, POPでレジスタ退避をする。

図1-2 レジスタ・バンクの使い分け例

<メイン・プログラム>



RBSを、サブルーチン処理あるいは割り込み処理で変更する場合は、PUSH/POP命令によって退避／復帰します。

RBEの設定は、SET1/CLR1命令によって行います。RBSの設定は、SEL命令により行います。

```

例 SET1 RBE ; RBE←1
      CLR1 RBE ; RBE←0
      SEL RB0 ; RBS←0
      SEL RB3 ; RBS←3
  
```

μ PD750008に内蔵されている汎用レジスタ・エリアは、4ビット・レジスタとしての使用のほかに、レジスタ・ペアによる8ビット・レジスタとして使用ができ、8ビット・マイコンに匹敵する転送、演算、比較、増減命令によって、汎用レジスタ中心のプログラミングが可能となります。

(1) 4ビット・レジスタとして使用する場合

汎用レジスタ・エリアを4ビット・レジスタとして使用する場合には、図1-3に示すように、 $RB = RBE \cdot RBS$ で指定されたレジスタ・バンクのX, A, B, C, D, E, H, L、計8個の汎用レジスタを使うことができます。このうちAレジスタは4ビット・アキュームレータとして、4ビット・データの転送、演算、比較などに中心的な働きをします。ほかの汎用レジスタは、アキュームレータとの転送、比較、増減ができます。

(2) 8ビット・レジスタとして使用する場合

汎用レジスタ・エリアを8ビット・レジスタとして使用する場合には、図1-4に示すように、
 RB=RBE・RBSで指定されたレジスタ・バンクのレジスタ・ペアをXA, BC, DE, HLとし、レジスタ・
 バンク(RB)のビット0を反転したレジスタ・バンクのレジスタ・ペアをXA', BC', DE', HL'として計
 8個の8ビット・レジスタを使うことができます。このうちXAレジスタ・ペアは、8ビット・アキュー
 ムレータとして、8ビット・データの転送、演算、比較などに中心的な働きをします。ほかのレジス
 タ・ペアは、アキュームレータとの転送、演算、比較、増減ができます。また、HLレジスタ・ペアは、
 主にデータ・ポインタとして機能します。DE, DLレジスタ・ペアも補助的なデータ・ポインタとして
 機能します。

例1. INCS HL ; HL←HL+1, HL=00Hでスキップ
 ADDS XA, BC ; XA←XA+BC, キャリーでスキップ
 SUBC DE', XA ; DE'←DE'-XA-CY
 MOV XA, XA' ; XA←XA'
 MOVT XA, @PCDE ; XA←(PC₁₂₋₈+DE) ROM, テーブル参照
 SKE XA, BC ; XA=BCならスキップ


2. タイマ/イベント・カウンタ0のカウント・レジスタ(T0)の値がBC'レジスタ・ペアの値
 より大きいかどうかをテストし大きくなるまで待つ。

CLR1 MBE ;
 NO: MOV XA, T0 ; カウント・レジスタ読み取り
 SUBS XA, BC' ; XA≥BC?
 BR YES ; YES
 BR NO ; NO

図1-3 汎用レジスタの構成（4ビット処理の場合）

X 01H	A 00H	レジスタ・バンク0 (RBE・RBS = 0)
H 03H	L 02H	
D 05H	E 04H	
B 07H	C 06H	
X 09H	A 08H	レジスタ・バンク1 (RBE・RBS = 1)
H 0BH	L 0AH	
D 0DH	E 0CH	
B 0FH	C 0EH	
X 11H	A 10H	レジスタ・バンク2 (RBE・RBS = 2)
H 13H	L 12H	
D 15H	E 14H	
B 17H	C 16H	
X 19H	A 18H	レジスタ・バンク3 (RBE・RBS = 3)
H 1BH	L 1AH	
D 1DH	E 1CH	
B 1FH	C 1EH	

図1-4 汎用レジスタの構成（8ビット処理の場合）



1.3 アプリケーション・プログラムの説明

このアプリケーション・ノートの各章では、応用プログラムを機能別にパッケージ化して記述しています。したがって、リンクを用いてユーザ・プログラム（メイン・プログラム）と組み合わせれば、システム・プログラムの一部として機能させることもできます。

また、第11章 このアプリケーション・プログラムの応用例では、第10章までに説明した応用プログラムのいくつかを使用し、1つのシステム・プログラムを作成しています。

各パッケージを使用するにあたっては、そのパッケージのプログラムの説明に従ってください。プログラムの説明の各項目の内容を次に示します。

- | | |
|---------------|--|
| 〈パブリック宣言シンボル〉 | : パッケージで使用するサブルーチンを示します。
このシンボルを外部参照宣言すると、そのパッケージ内で参照することができます。 |
| 〈外部参照宣言シンボル〉 | : パッケージで使用するデータ・メモリの領域を示します。
このシンボルを定義し、パブリック宣言すると、そのパッケージ内で使用することができます。 |
| 〈使用するレジスタ〉 | : パッケージで使用するレジスタを示します。 |
| 〈使用するRAM〉 | : パッケージで使用するデータ・メモリの領域を示します。 |
| 〈ネスティング〉 | : パッケージで許可されたネスティング・レベルを示します。
() 内の値は使用するスタックの最大値を示します。 |
| 〈使用するハードウェア〉 | : パッケージで使用するハードウェアを示します。 |
| 〈割り込み〉 | : パッケージで使用する割り込みを示します。 |
| 〈初期設定〉 | : パッケージを動作させるために必要な初期設定を示します。
ただし、各パッケージでは、特に記述がない場合はSCC = 0, PCC = 3となっています。 |
| 〈起動方法〉 | : パッケージを動作させるために必要な処置を示します。 |

各パッケージは、MkII モードで動作します。レジスタ・バンクは次の各割り込みで切り替えられます。

- | | |
|------------------|---------------------|
| RBE = 0, RBS = 0 | : メイン・ルーチン |
| RBE = 1, RBS = 1 | : INTO割り込み（リモコンの受信） |
| RBE = 1, RBS = 2 | : ベーシック・インターバル・タイマ |
| RBE = 1, RBS = 3 | : SBI |

次に、プログラムの記述例を示します。

```

VENT0  MBE = 0, RBE = 0, INIT      ;RESET
VENT1  MBE = 0, RBE = 1, INTBT    ;INTBT/INT4
VENT2  MBE = 0, RBE = 1, INTO     ;INTO
VENT4  MBE = 0, RBE = 1, SBI      ;SBI

;*****
; 初期設定
;*****

INIT:
    DI                      ;すべての割り込みを禁止
    CLR1      MBE           ;MBE←0
    MOV        XA, #00H       ;
    MOV        SP, XA         ;STACK・ポインタ←00H
    MOV        SBS, A         ;メモリ・バンク←0, Mk II モードを選択

;*****
; リモコン処理
;*****


INTO:
    DI                      ;
    PUSH     BS             ;
    SEL      RB1            ;レジスタ・バンク←1

;*****
; タイマ
;*****


INTBT:
    SEL      RB2            ;レジスタ・バンク←2

;*****
; SBI処理
;*****


SBI:
    SEL      RB3            ;レジスタ・バンク←3

```

第2章 システム・クロック切り替え機能の応用

μ PD750008は、プロセッサ・クロック・コントロール・レジスタ（PCC）とシステム・クロック・コントロール・レジスタ（SCC）を書き換えることにより、CPUクロックおよびシステム・クロックを切り替えることが可能です。図2-1にPCC、図2-2にSCCのフォーマットを示します。

図2-1 プロセッサ・クロック・コントロール・レジスタのフォーマット

アドレス	3	2	1	0	略号																																									
FB3H	PCC3	PCC2	PCC1	PCC0	PCC																																									
CPUクロック選択ビット (fx = 6.0 MHz時の場合)																																														
<table border="1" style="width: 100%; border-collapse: collapse;"> <thead> <tr> <th colspan="2"></th> <th colspan="2">SCC3, SCC0 = 00</th> <th colspan="2">SCC3, SCC0 = 01 or 11</th> </tr> <tr> <th colspan="2"></th> <th colspan="2">() 内はfx = 6.0 MHz時</th> <th colspan="2">() 内はfx_T = 32.768 kHz時</th> </tr> <tr> <th colspan="2"></th> <th>CPUクロック周波数</th> <th>1マシン・サイクル</th> <th>CPUクロック周波数</th> <th>1マシン・サイクル</th> </tr> </thead> <tbody> <tr> <td>0</td> <td>0</td> <td>$\Phi = fx/64$ (93.7 kHz)</td> <td>10.7 μs</td> <td>$\Phi = fx_T/4$ (8.192 kHz)</td> <td>122 μs</td> </tr> <tr> <td>0</td> <td>1</td> <td>$\Phi = fx/16$ (375 kHz)</td> <td>2.67 μs</td> <td></td> <td></td> </tr> <tr> <td>1</td> <td>0</td> <td>$\Phi = fx/8$ (750 kHz)</td> <td>1.33 μs</td> <td></td> <td></td> </tr> <tr> <td>1</td> <td>1</td> <td>$\Phi = fx/4$ (1.5 MHz)</td> <td>0.67 μs</td> <td></td> <td></td> </tr> </tbody> </table>							SCC3, SCC0 = 00		SCC3, SCC0 = 01 or 11				() 内はfx = 6.0 MHz時		() 内はfx _T = 32.768 kHz時				CPUクロック周波数	1マシン・サイクル	CPUクロック周波数	1マシン・サイクル	0	0	$\Phi = fx/64$ (93.7 kHz)	10.7 μ s	$\Phi = fx_T/4$ (8.192 kHz)	122 μ s	0	1	$\Phi = fx/16$ (375 kHz)	2.67 μ s			1	0	$\Phi = fx/8$ (750 kHz)	1.33 μ s			1	1	$\Phi = fx/4$ (1.5 MHz)	0.67 μ s		
		SCC3, SCC0 = 00		SCC3, SCC0 = 01 or 11																																										
		() 内はfx = 6.0 MHz時		() 内はfx _T = 32.768 kHz時																																										
		CPUクロック周波数	1マシン・サイクル	CPUクロック周波数	1マシン・サイクル																																									
0	0	$\Phi = fx/64$ (93.7 kHz)	10.7 μ s	$\Phi = fx_T/4$ (8.192 kHz)	122 μ s																																									
0	1	$\Phi = fx/16$ (375 kHz)	2.67 μ s																																											
1	0	$\Phi = fx/8$ (750 kHz)	1.33 μ s																																											
1	1	$\Phi = fx/4$ (1.5 MHz)	0.67 μ s																																											
(fx = 4.19 MHz時の場合)																																														
<table border="1" style="width: 100%; border-collapse: collapse;"> <thead> <tr> <th colspan="2"></th> <th colspan="2">SCC3, SCC0 = 00</th> <th colspan="2">SCC3, SCC0 = 01 or 11</th> </tr> <tr> <th colspan="2"></th> <th colspan="2">() 内はfx = 4.19 MHz時</th> <th colspan="2">() 内はfx_T = 32.768 kHz時</th> </tr> <tr> <th colspan="2"></th> <th>CPUクロック周波数</th> <th>1マシン・サイクル</th> <th>CPUクロック周波数</th> <th>1マシン・サイクル</th> </tr> </thead> <tbody> <tr> <td>0</td> <td>0</td> <td>$\Phi = fx/64$ (65.5 kHz)</td> <td>15.3 μs</td> <td>$\Phi = fx_T/4$ (8.192 kHz)</td> <td>122 μs</td> </tr> <tr> <td>0</td> <td>1</td> <td>$\Phi = fx/16$ (261.8 kHz)</td> <td>3.82 μs</td> <td></td> <td></td> </tr> <tr> <td>1</td> <td>0</td> <td>$\Phi = fx/8$ (524 kHz)</td> <td>1.91 μs</td> <td></td> <td></td> </tr> <tr> <td>1</td> <td>1</td> <td>$\Phi = fx/4$ (1.05 MHz)</td> <td>0.95 μs</td> <td></td> <td></td> </tr> </tbody> </table>							SCC3, SCC0 = 00		SCC3, SCC0 = 01 or 11				() 内はfx = 4.19 MHz時		() 内はfx _T = 32.768 kHz時				CPUクロック周波数	1マシン・サイクル	CPUクロック周波数	1マシン・サイクル	0	0	$\Phi = fx/64$ (65.5 kHz)	15.3 μ s	$\Phi = fx_T/4$ (8.192 kHz)	122 μ s	0	1	$\Phi = fx/16$ (261.8 kHz)	3.82 μ s			1	0	$\Phi = fx/8$ (524 kHz)	1.91 μ s			1	1	$\Phi = fx/4$ (1.05 MHz)	0.95 μ s		
		SCC3, SCC0 = 00		SCC3, SCC0 = 01 or 11																																										
		() 内はfx = 4.19 MHz時		() 内はfx _T = 32.768 kHz時																																										
		CPUクロック周波数	1マシン・サイクル	CPUクロック周波数	1マシン・サイクル																																									
0	0	$\Phi = fx/64$ (65.5 kHz)	15.3 μ s	$\Phi = fx_T/4$ (8.192 kHz)	122 μ s																																									
0	1	$\Phi = fx/16$ (261.8 kHz)	3.82 μ s																																											
1	0	$\Phi = fx/8$ (524 kHz)	1.91 μ s																																											
1	1	$\Phi = fx/4$ (1.05 MHz)	0.95 μ s																																											
備考 1. fx: メイン・システム・クロック発振回路出力周波数																																														
2. fx _T : サブシステム・クロック発振回路出力周波数																																														
CPU動作モード制御ビット																																														
<table border="1" style="width: 100%; border-collapse: collapse;"> <tbody> <tr> <td>0</td> <td>0</td> <td>通常動作モード</td> </tr> <tr> <td>0</td> <td>1</td> <td>HALTモード</td> </tr> <tr> <td>1</td> <td>0</td> <td>STOPモード</td> </tr> <tr> <td>1</td> <td>1</td> <td>設定禁止</td> </tr> </tbody> </table>						0	0	通常動作モード	0	1	HALTモード	1	0	STOPモード	1	1	設定禁止																													
0	0	通常動作モード																																												
0	1	HALTモード																																												
1	0	STOPモード																																												
1	1	設定禁止																																												

図2-2 システム・クロック・コントロール・レジスタのフォーマット





図2-3に電源電圧 (V_{DD}) に対する最小命令実行時間 (t_{CY}) を示します。

図2-3 電源電圧 (V_{DD}) に対する最小命令実行時間 (t_{CY})

2.1 RESET後のPCCの切り替え

RESET信号発生により、CPUクロックはメイン・システム・クロックの最低速モードが選択されます。したがって、高速処理を行う場合には、PCCを書き換えてCPUクロックを最高速モードとします。ただしPCCを書き換える以前に、V_{DD}端子電圧が高速で動作できる電圧まで、上昇していかなければなりません（図2-3参照）。またサブシステム・クロックを使用するシステムではあらかじめ、サブシステム・クロックの発振が開始していることを確認しておく必要があります。

ここでは、時計用タイマを用いて、V_{DD}端子電圧が上昇するまでのウエイトおよびサブシステム・クロックの発振の確認を行い、CPUクロックを最高速に切り替えるプログラム例（f_x = 4.19 MHz, f_{XT} = 32.768 kHz時）を示します。また図2-4にそのタイミング図を示します。

- プログラム例

（1）V_{DD}電圧上昇までのウエイト

```

EXTRN BIT(SSCOKF)      ;サブシステム・クロック発振確認フラグ(初期値 = 0)
VENTO MBE=0, RBE=0, START
;,,,,,,,,;,,,,,,,,;,,,,,,,,;,,,,,,,,;,
START CSEG INBLOCK      ;プログラム・メモリの4Kバイトのブロック内に配置することを指定
    MOV XA, #00000110B   ;時計用タイマ早送りモード
    MOV WM, XA            ;(メイン・システム・クロックで動作)
INIOP1:
    SKTCLR IRQW          ;3.9 msウエイト注
    BR INIOP1
    MOV A, #0010B          ;ドライブ電流少
    MOV SOS, A
    MOV XA, #00000110B    ;時計用タイマ・サブシステム・クロックで動作
    MOV WM, XA
    MOV A, #0011B          ;CPUクロック最高速
    MOV PCC, A              ;以下イニシャライズ処理

```

注 この時間はハードウェアに依存します。各システムに合った時間を設定してください。


(2) サブシステム・クロック発振確認サブルーチン（ユーザ処理にてコールします）

```

SSCCHK CSEG INBLOCK
SKF     SSCOKF      ;すでに1回サブシステム・クロックの発振を確認した場合は何もしない
RET
SKTCLR IROW        ;サブシステム・クロック発振
RET
MOV     XA, #00000100B ;時計用タイマ通常動作モード
MOV     WM, XA
SET1   SSCOKF      ;サブシステム・クロック発振確認フラグをセット
RET
END

```

サブシステム・クロックは、発振が安定するまでに約1sかかるので、イニシャライズ処理内で発振が安定するまでウエイトせずに、メイン・ルーチン内でサブシステム・クロック発振確認の処理を行い、フラグをセットすることによって、サブシステム・クロックが発振を開始していることを他の処理に知らせます。

図2-4 $\overline{\text{RESET}}$ 後のCPUクロック切り替え

注1. マスク・オプションにより、選択することができます（4.19 MHz動作時は、 $2^{15}/f_x = 7.81\text{ ms}$, $2^{17}/f_x = 31.3\text{ ms}$ ）。

2. この時間はハードウェアに依存します。各システムに合った時間を設定してください。

2.2 商用電源の停電検出時のシステム・クロックの切り替え

μ PD750008は、SCCの設定によりサブシステム・クロック(32.768 kHz)を選択し、超低消費電流で動作させることができます。したがってNiCd電池やスーパー・キャパシタなどのバックアップ用電源をシステム上に付加することにより、停電時に、時計カウントなどの処理を超低消費電力で継続することができます。

ここでは、商用電源のオン／オフを外部割り込みINT4により検出し、システム・クロックを切り替え、超低消費電力で時計機能を継続するプログラム例を示します。

図2-5および図2-6に従ってシステム・クロックの切り替え手順を説明します。

図2-5 商用電源オン／オフ時のシステム・クロックの切り替え





図2-6 INT4割り込み処理のアルゴリズム



2.2.1 パワーオフ処理

INT4割り込み処理の中でPort0.0をチェックし、ロウ・レベルのときは商用電源オフと判断し、パワーオフ処理を行います。

次にパワーオフ処理の処理手順を示します。

- ① PCCに0011Bを設定します。
- ② システム・クロックをサブシステム・クロックにするためSCCのビット0をセットします。
システム・クロックが変化する前に時計用タイマとベーシック・インターバル・タイマ（停止することができない）を除く周辺ハードウェアを停止させます。
- ③ 入出力端子を消費電流が最小となるように処理します。
- ④ SCCのビット0をセットしてから32マシン・サイクル以上経過後、SCCのビット3をセットし、メイン・システム・クロックの発振を停止させます。

注意 パワーオフ時の端子の処理について

商用電源がオフになった場合、通常周辺回路の電源はオフになります。そのため、接続されている周辺回路により、次のような処置が必要です。

1. 端子に接続されている周辺回路がハイ・インピーダンスになる場合
 - ・入出力が切り替えられる端子は出力モードとし、ロウ・レベルを出力します。
 - ・入力端子は、あらかじめプルダウンまたはプルアップしておきます。
2. 端子に接続されている周辺回路がハイ・インピーダンスにならない場合
 - ・出力端子は電流が流れないレベルを出力します。
 - ・入力端子は入力がハイ・レベルまたはロウ・レベルで安定していれば、特に処置をする必要はありません。

その他、周辺回路の状況により必要な処置を行ってください。

2.2.2 パワーオン処理

INT4割り込み処理の中で、Port0.0をチェックし、ハイ・レベルのときは商用電源オンと判断し、パワーオン処理を行います。

次にパワーオン処理の処理手順を示します。

- ① VDD端子電圧が最高速で動作できる電圧に上昇するまでウエイトします。
- ② SCCのビット3をクリアしてメイン・システム・クロックの発振を開始します。
- ③ メイン・システム・クロックの発振が安定するのに必要な時間だけウエイトし、SCCのビット0をクリアして、システム・クロックを切り替えます。

2.2.3 パワーオン／オフ処理の応用

パワーオン／オフ処理を使用したプログラムの例を次に示します。

このプログラムは、第11章 このアプリケーション・プログラムの応用例には使用されていません。

(1) プログラムの説明

〈外部参照宣言シンボル〉

SSCOKF : サブシステム・クロック発振確認フラグ

INTD0-5 : 割り込み関係イニシャライズ・データ

INDBTM : BTMイニシャライズ・データ

INTM0 : TM0イニシャライズ・データ

INCSIM : CSIMイニシャライズ・データ

INWM : WMイニシャライズ・データ

WATCH : 時計カウント処理エントリ・ラベル

〈使用するレジスタ〉

XA, HL

〈ネスティング〉

4 レベル (20ワード)

〈初期設定〉

INT4割り込み許可

〈起動方法〉

(a) パワーオン／オフ処理

INT4割り込みにより起動したとき、Port0.0の状態により次の2つの処理に分かれます。

(i) Port0.0がハイ・レベルの場合 (パワーオン処理)

システム・クロックをメイン・システム・クロックに切り替え、各周辺ハードウェアを起動します。

(ii) Port0.0がロウ・レベルの場合 (パワーオフ処理)

- サブシステム・クロックが発振しているとき (SSCOKF = 1)

時計用タイマ、ベーシック・インターバル・タイマを除く周辺ハードウェアを停止します。その後、システム・クロックをサブシステム・クロックに切り替え、メイン・システム・クロックを停止します。

- ・サブシステム・クロックが停止しているとき ($SSCOKF = 0$)
時計用タイマ, ベーシック・インターバル・タイマを除く周辺ハードウェアを停止します。

(b) 時計カウント・チェック・サブルーチン

`TIMCNT`をコールすると、サブシステム・クロックが発振しているときには時計カウントの処理ルーチン（ユーザ・プログラム）をコールします。サブシステム・クロックが停止しているときは、時計カウントの処理ルーチンをコールせずにリターンします。

また、パワーオフ（`Port0.0`がロウ・レベル）の間は次の処理を繰り返し、パワーオン（`Port0.0`がハイ・レベル）になったときにリターンします。

- (i) HALTモードの設定
- (ii) `IRQW`により復帰
- (iii) 時計カウントの処理ルーチンのコール

(2) プログラム例

```

<INT4割り込み処理>
    VENT1      MBE=0, RBE=0, INT4
    EXTRN      NUMBER(INTD0, INTD1, INTD2, INTD3, INTD4, INTD5)
    EXTRN      NUMBER(INDBTM, INTM0, INC SIM, INWM)
    EXTRN      BIT(SSCOKF)
    EXTRN      CODE(WATCH)
AN12DT1  DSEG      0          AT 0
ROA:      DS        1H          ; A レジスタ
AN12DT   DSEG      0          AT 0DOH
PTSVA:    DS        4
;
;
INT4     CSEG      INBLOCK
PUSH      BS
PUSH      XA
PUSH      HL
SEL       MB15
SKT       PORT0.0      ;Port0.0チェック
BR       POWOFF
CLR1     SCC.3      ;パワーオン処理
MOV       XA, #56
ONWAIT:
DECS      A          ;システム・クロックの発振が安定するまでウェイト
BR       ONWAIT
DECS      X
BR       ONWAIT
CLR1     SCC.0      ;システム・クロックの切り替え
CALLF    !HRDSET    ;ハードウェア再起動サブルーチン
BR       REINT4
POWOFF:
SKT       SSCOKF    ;サブシステム・クロックの発振確認
BR       $HSTPRO
MOV       A, #0011B
MOV       PCC, A
SET1     SCC.0      ;システム・クロックの切り替え
HSTPRO:
MOV       XA, #0      ;ハードウェア停止処理
MOV       TMO, XA
MOV       CSIM, XA
MOV       A, #0101B
SKF       SSCOKF
MOV       WM, XA
MOV       A, #8      ;INT4以外の割り込みを禁止
MOV       OB8H, A
SET1     MBE
MOV       HL, #0BCH
MOV       A, #0
DIOP:
MOV       @HL, A
INCS      L
BR       DIOP
CLR1     MBE
EI       IEW

```

```

; <サブルーチン 入出力ポート処理>
SKF      SSCOKF
SET1    SCC.3      ;メイン・システム・クロックを停止
REINT4:
POP     HL
POP     XA
POP     BS
RETI

; <サブルーチン ハードウェア再起動>
HRDSET CSEG   SENT
MOV     A, #INDBTM    ;ハードウェアを再起動
MOV     BTM, A
MOV     XA, #INTMO
MOV     TMO, XA
MOV     XA, #INCSEL
MOV     CSIM, XA
MOV     XA, #INWM
MOV     WM, XA
;

;

;

PORT RECOVER PROCESS
;

;割り込みを許可
;

MOV     A, #INTD0
MOV     OB8H, A
MOV     A, #INTD1
SKT     ROA.1
DI      IEW
MOV     A, #INTD2
MOV     OBCH, A
MOV     A, #INTD3
MOV     OBDH, A
MOV     A, #INTD4
MOV     OBEH, A
MOV     A, #INTD5
MOV     OBFH, A
RET

; <サブルーチン 時計カウント・チェック>
TIMCNT CSEG   INBLOCK
SKF     SSCOKF    ;サブシステム・クロック、チェック中は時計カウントを行わない
CALL    !WATCH
SKF     PORT0.0    ;パワーダウンのチェック
RET
HALT
NOP
BR     TIMCNT    ;スタンバイ・モードの設定

```

第3章 ベーシック・インターバル・タイマの応用


3.1 基準時間発生

3

μ PD750008は8ビットのベーシック・インターバル・タイマ(BT)を備えています。このベーシック・インターバル・タイマは4種類のインターバル時間を設定することができ、このインターバル時間ごとにベーシック・インターバル・タイマ割り込み要求フラグ(IRQBT)をセットします。

ベーシック・インターバル・タイマは、ベーシック・インターバル・タイマ・モード・レジスタ(BTM)によって制御されます。BTMのフォーマットを図3-1に示します。

図3-1 ベーシック・インターバル・タイマ・モード・レジスタのフォーマット



次に4種類のインターバル時間を設定した例を示します（すべて $f_x = 4.19\text{ MHz}$ 動作時）。

(1) インターバル時間を250 msに設定する（250 msごとにIRQBTをセットする）。

```
SEL MB15  
MOV A, #1000B  
MOV BTM, A
```

(2) インターバル時間を31.3 msに設定する（31.3 msごとにIRQBTをセットする）。

```
SEL MB15  
MOV A, #1011B  
MOV BTM, A
```

(3) インターバル時間を7.81 msに設定する（7.81 msごとにIRQBTをセットする）。

```
SEL MB15  
MOV A, #1101B  
MOV BTM, A
```

(4) インターバル時間を1.95 msに設定する（1.95 msごとにIRQBTをセットする）。

```
SEL MB15  
MOV A, #1111B  
MOV BTM, A
```

3.2 ウオッチドッグ・タイマの応用

ベーシック・インターバル・タイマ／ウォッチドッグ・タイマは、ウォッチドッグ・タイマ許可フラグ (WDTM) に“1”をセットすると、ベーシック・インターバル・タイマ (BT) のオーバフローにより内部リセット信号を発生するウォッチドッグ・タイマとして動作します（なお、WDTMは一度“1”にセットすると、リセット以外にクリアすることはできません）。BTは、クロック発生回路からのクロックによって常にインクリメントされ、カウント動作を停止することはできません（図3-2参照）。

ウォッチドッグ・タイマ・モードでは、BTのオーバフローするインターバル時間を利用して、プログラムの暴走を検出します。このインターバル時間は、BTMのビット2-0の設定により4通りの時間が選択できます（図3-1参照）。これらの中からユーザ・システムに応じて暴走検出に必要な時間を決めてください。インターバル時間を設定しておいて、プログラムをその時間内に実行できる単位に分割し、それぞれの単位の最後でBTをクリアする命令を実行させるようにします。そうすると、設定時間内にこのBTクリアを実行する命令にたどりつかなければ（順調にプログラムの実行が進んでいなければ=暴走）BTはオーバフローし、内部リセット信号が発生してプログラムを強制終了させてしまいます。この結果、内部リセットがかかったということはプログラムの暴走が起きたことを示し、その検出ができたことになります。

ウォッチドッグ・タイマの設定手順は、次のように行ってください（①、②の設定は同時にあってもかまいません）。

- ① BTMにインターバル時間をセットする。
- ② BTMのビット3に“1”をセットする。 } 初期設定
- ③ WDTMに“1”をセットする。 }
- ④ ①～③を設定したあとは、インターバル時間以内にBTMのビット3に“1”をセットする。

図3-2 ウォッチドッグ・タイマ許可フラグのフォーマット

アドレス	
F8BH.3	
0	BTモード ベーシック・インターバル・タイマ (BT) のオーバフローによりIRQBTをセットします。
1	WTモード ベーシック・インターバル・タイマ (BT) のオーバフローにより内部リセット信号を発生します。

例 7.81 msのウォッチドッグ・タイマとして使用する（4.19 MHz動作時）。

プログラムをBTMの設定時間（7.81 ms）以内に処理が終了するいくつかのモジュールに分割し、各モジュールの終わりでBTをクリアする。暴走した場合、BTが設定時間内にクリアされないためオーバーフローしてしまい、内部リセット信号が発生する。

初期設定：

```

SET1    MBE
SEL     MB15
MOV    A,#1101B
MOV    BTM,A      ; 時間設定とスタート
SET1    WDTM      ; ウォッチドッグ・タイマを許可
:

```

（以後、7.81 msごとにBTMのビット3に“1”をセットする。）

モジュール1：

```

:
SET1    MBE
SEL     MB15
SET1    BTM.3

```

↑
↓
7.81 ms以内で
処理完了

モジュール2：

```

:
SET1    MBE
SEL     MB15
SET1    BTM.3

```

↑
↓
7.81 ms以内で
処理完了


3.3 リモコン信号受信応用

ベーシック・インターバル・タイマを使用して、汎用赤外線リモート・コントローラ（以後、リモコンと略します）送信用デバイス μ PD6122の送信データを受信するプログラムの例を紹介します。

リモコン信号はPIN受光ダイオードで受信され、リモコン用受信プリアンプ μ PC2800Aを介して、P10/INT0端子から入力されます（図3-3参照）。

このプログラムは、そのリモコン信号のエッジ間の長さをINTBTでカウントして、コード化します。


図3-3 リモコン信号受信回路例



リモコン信号は、リーダ・コード（Leader Code）、カスタム・コード（Custom Code）、データ・コード（Data Code）、リピート・コード（Repeat Code）からなります。

ここで示されるリモコン信号は、NECのリモコン信号フォーマットに準拠しています。そのフォーマットを図3-4に示します。

図3-4 リモコン送信用IC出力信号




リモコン用受信プリアンプ μ PC2800Aの出力は、図3-5 (a) のように、口ウ・アクティブです。


しかし、実際には外来光（蛍光灯など）のノイズがリーダ・コードの前に多数現れます。そこでこのプログラムでは、リーダ・コードの立ち下がりの確認は、ベーシック・インターバル・タイマの割り込み処理内で、ポートのレベルの読み出しにより行います（図3-5 (b) 参照）。

図3-5 受信プリアンプの出力波形

(a) 理想の波形



(b) 実際の波形



次の（1）プログラムの説明と（2）フロー・チャートでは、リモコン信号で正常と認識されたデータを“有効な”と表現しています。

(1) プログラムの説明

このプログラムは、第11章 このアプリケーション・プログラムの応用例で使用するものを、入出力インターフェース部を変更して説明しています。

〈使用するレジスタ〉

XA, B, HL

〈使用するRAM〉

RAMは、メモリ・バンク0の20H番地から配置できます。

WORK :	2ワード	; 受信したデータを格納するワーク・エリア
RMDATA :	2ワード	; 有効なデータ・コードを格納するエリア
LDCODE :	1ワード	; リーダ・コードのロウ・レベル時間をカウントする
MODEP :	1ワード	; どのエッジを検出中かを示すパラメータ
RPTIM :	1ワード	; 有効なデータ入力から検出した時間をカウントする (200 msまで)
RPCODE :	1ワード	; 有効なリピート・コードをカウントする
RMFLG :	1ワード	; フラグを格納する
REP_F :	1ビット	; 1ならばRPTIMのカウントをスタートする

〈ネスティング〉

2 レベル (16ワード)

〈使用するハードウェア〉

- ・ポート：Port1.0 (INT0として使用)
- ・タイマ：ベーシック・インターバル・タイマ

〈使用する割り込み〉

INT0, INTBT

〈初期設定〉


- ・RAMの初期設定
- ・ハードウェアの初期設定
- ・INTBTの割り込み許可


〈起動方法〉

- ・初期設定処理 (RMT:) の実行により、起動します。
- ・有効なリモコン・コードが入力されると、REP_F = 1とし、RMDATAにコードが格納されます。その後200 msの間に1回もリピート・コードが入力されなかった場合（リモコンのキーが離されたと判定）、また4回以上リピート・コードが入力された場合（2機以上のリモコン信号を受けたと判定）はREP_F = 0とします。
- ・リモコン・コードの受信を待機している間はMODEP = 0となります。


(2) フロー・チャート


初期設定







INTBT処理









(3) プログラム例

```

VENT0      MBE=0, RBE=0, INIT      ;RESET
VENT1      MBE=0, RBE=1, INTBT     ;INTBT/INT4
VENT2      MBE=0, RBE=1, INTO       ;INTO

DSEG0    DSEG    0      AT      20H      ;メモリ・バンク0, アドレス20Hから格納

WORK:        DS      2      ;受信したデータを格納するワーク・エリア
RMDATA:      DS      2      ;有効なデータ・コードを格納
LDCODE:      DS      1      ;リーダ・コードのロウ・レベル時間をカウント
MODEP:        DS      1      ;どのエッジを検出中かを示すパラメータ
RPTIM:        DS      1      ;有効なデータ入力からの経過時間をカウント(200msまで)
RPCODE:      DS      1      ;有効なリピート・コードをカウント
RMFLG:        DS      1      ;フラグ

REP_F      EQU      RMFLG.0      ;1ならRPTIMカウント・スタート

; <サブルーチン 初期設定>

INIT:
    DI
    CLR1      MBE      ;すべての割り込みを禁止
                  ;MBE←0

    MOV      XA, #00H      ;RAMを設定
    MOV      WORK, XA
    MOV      RMDATA, XA
    MOV      MODEP, A
    MOV      RPTIM, A
    MOV      RPCODE, A
    MOV      RMFLG, A
    MOV      A, #0DH
    MOV      LDCODE, A      ;LDCODEを初期化

    MOV      A, #0011B
    MOV      PCC, A      ;1マシン・サイクル←0.95μs, CPUは通常動作モードにする

    MOV      A, #1111B
    BTM, A      ;インターバル時間←1.95ms

    EI
    EI      IEBT      ;ベーシック・インターバル・タイマ(INTBT)の使用を許可
                  ;すべての割り込みを許可

```

; <サブルーチン 100 msタイマ処理>

SKT	REP_F	
RET		
INCS	RPTIM	;リモコン・リピート用タイマ+1
NOP		
MOV	A, RPTIM	
SKE	A, #2H	;リモコン・リピート用タイマ=200ms?
RET		
MOV	A, #0H	
MOV	RPTIM, A	;RPTIMをクリア
MOV	A, RPCODE	
ADDS	A, #0FH	;リモコン・リピート・コード・カウンタ=0?
BR	TIM_52	
ADDS	A, #0FH	;リモコン・リピート・コード・カウンタ=1?
RET		
ADDS	A, #0FH	;リモコン・リピート・コード・カウンタ=2?
RET		
ADDS	A, #0FH	;リモコン・リピート・コード・カウンタ=3?
RET		
TIM_52:		
CLR1	REP_F	;REP_F←0
RET		

; <サブルーチン INTBT処理>

```

SEL      RB2           ;レジスタ・バンク←2
SKTCLR  IRQT1         ;1ならば、モジュロ・レジスタ=カウント・レジスタ
BR       MODEP7        ;
MOV      A, #0H         ;
MOV      MODEP, A      ;MODEP←0H
MOV      A, #0DH         ;
MOV      LDCODE, A      ;LDCODEを初期化
DI       IEO            ;INT0割り込みを禁止
BR       RMCNT          ;

MODEP7:
MOV      A, MODEP        ;
SKE      A, #7H          ;MODEP=7?
BR       RMCNT          ;
MOV      A, #0H          ;
MOV      MODEP, A        ;MODEP←0H
MOV      RPCODE, A      ;リピート・コード・カウント・リセット
MOV      RPTIM, A        ;200msカウント・リセット

RMCNT:
MOV      A, MODEP        ;
SKE      A, #0H          ;MODEP=0?
RETI    PORT1.0         ;
BR       RMCNT_1         ;
INCS   LDCODE          ;リーダ・コード・スキャン・カウンタ+1
RETI    PORT1.0         ;
MOV      A, #0DH          ;LDCODEを初期化
MOV      LDCODE, A      ;LDCODE←0H
MOV      A, #1H          ;
MOV      MODEP, A        ;MODEP←1
MOV      A, #0H          ;
MOV      IMO, A          ;立ち上がりエッジを指定
EI       IEO            ;INT0割り込みを許可
RETI    PORT1.0         ;

RMCNT_1:
MOV      A, #0DH          ;LDCODEを初期化
MOV      LDCODE, A      ;
RETI    PORT1.0         ;

```

; <サブルーチン INT0処理>

```

DI
PUSH    BS
SEL     RB1           ;レジスタ・バンク←1
MOV     A, MODEP
ADDS   A, #0FH        ;MODEP=0?
BR     INTO_E
ADDS   A, #0FH        ;MODEP=1?
BR     INTO_P1
ADDS   A, #0FH        ;MODEP=2?
BR     INTO_P2
ADDS   A, #09H        ;MODEP=3-6?
BR     INTO_PX
BR     INTO_E

INT0_P1:
MOV     A, #2H
MOV     MODEP, A       ;MODEP← 2
MOV     XA, #62H
MOV     TMOD1, XA      ;モジュロ・レジスタ←6ms
MOV     A, #1H
MOV     IMO, A          ;立ち下がりエッジ指定
BR     INTO_E

INT0_P2:
MOV     XA, T1
ADDS   XA, #0C9H       ;カウント・レジスタ=3.375ms?
BR     INTO_RP1
MOV     A, #3H
MOV     MODEP, A        ;MODEP← 3
BR     INTO_W1

INT0_RP1:
INCS   RPCODE         ;リモコン・リピート・コード・カウンタ+1
NOP
BR     INTO_P0

INT0_PX:
MOV     XA, T1
ADDS   XA, #0E4H       ;カウント・レジスタ=1.6875ms?
BR     INTO_CY0
SET1   CY              ;CY← 1
BR     INTO_L1

INT0_CY0:
CLR1   CY              ;CY← 0

INT0_L1:
MOV     XA, WORK
MOV     B, A
MOV     A, X
RORC   A
MOV     X, A
MOV     A, B
RORC   A

```

```

MOV     WORK, XA           ;受信データ(WORK)を右1ビット・シフト
SKT     CY
BR     INTO_E
MOV     A, MODEP
ADDS   A, #0CH             ;MODEP=3?
BR     INTO_P3
ADDS   A, #0FH             ;MODEP=4?
BR     INTO_P4
ADDS   A, #0FH             ;MODEP=5?
BR     INTO_P5
MOV     XA, WORK            ;MODEP=6
MOV     HL, #0FFH
XOR     HL, XA
MOV     XA, RMDATA
SKE     XA, HL             ;有効なりモコン・データ=受信データを反転?
BR     INTO_P0
MOV     A, #7H
MOV     MODEP, A            ;MODEP←7
SET1   REP_F
DI      IEO                ;INT0割り込み禁止
MOV     XA, #0H
MOV     TM1, XA             ;タイマ・カウントを停止
BR     INTO_E1

INT0_P3:
MOV     XA, WORK
MOV     HL, #00H
SKE     XA, HL             ;カスタム・コード=00H?
BR     INTO_P0
MOV     A, #4H
MOV     MODEP, A            ;MODEP←4
BR     INTO_W1

INT0_P4:
MOV     XA, WORK
MOV     HL, #0FFH
SKE     XA, HL             ;カスタム・コード=FFH?
BR     INTO_P0
MOV     A, #5H
MOV     MODEP, A            ;MODEP←5
BR     INTO_W1

INT0_P5:
MOV     XA, WORK
MOV     RMDATA, XA          ;有効なりモコン・データに受信データを入力
MOV     A, #6H
MOV     MODEP, A            ;MODEP←6
BR     INTO_W1

INT0_P0:
MOV     A, #0H
MOV     MODEP, A            ;MODEP←0
MOV     A, #0DH
MOV     LDCODE, A           ;LDCODEを初期化
DI      IEO                ;INT0割り込みを禁止
MOV     XA, #0H

```

```
MOV      TM1, XA          ;タイマ・カウント停止
BR      INTO_E1

INTO_W1:
MOV      XA, #0FFH
MOV      TMOD1, XA        ;モジュロ・レジスタ←4ms
MOV      XA, #80H
MOV      WORK, XA         ;WORKを初期化

INTO_E:
MOV      A, MODEP
ADDS   A, #0DH
MOV      XA, #6CH
MOV      XA, #7CH
MOV      TM1, XA          ;タイマ・カウント・スタート

INTO_E1:
POP     BS
EI
RETI  ;すべての割り込みを許可
```


表 4-1 分解能と最長設定時間 ($f_x = 4.194304 \text{ MHz}$)

モード・レジスタ			分解能	最長時間
TM06	TM05	TM04		
1	0	0	244 μs	62.5 ms
1	0	1	61 μs	15.6 ms
1	1	0	15.3 μs	3.9 ms
1	1	1	3.8 μs	977 μs

次にタイマ・モード・レジスタとモジュロ・レジスタによってインターバル時間を設定した例を示します。

例1、2では、インターバル時間ごとにタイマ／イベント・カウンタ割り込み要求フラグ (IRQT0) をセットします。

例1. 10 msのインターバル時間を設定する ($f_x = 4.194304 \text{ MHz}$)。

最長設定時間が10 ms以上で、分解能が最も小さいモードを使用します。これは表 4-1 より最長設定時間が15.6 msのモードとなります。

モジュロ・レジスタに設定する値は、②式により次のようにになります。

$$N = \frac{T}{(\text{分解能})} - 1$$

$$= \frac{10 \times 10^{-3}}{61 \times 10^{-6}} - 1$$

$$= 162.9 = 0A3H$$

TIMER0:

```

SEL      MB15
MOV      XA, #0A3H
MOV      TMOD0, XA          ;モジュロ・レジスタを設定
MOV      XA, #01011100B
MOV      TMO, XA           ;タイマ・モード・レジスタを設定

```

例2. 120 μ sのインターバル時間を設定する ($f_x = 4.194304 \text{ MHz}$)。

最長設定時間が120 μ s以上で、分解能が最も小さいモードを使用します。これは表4-1より最長設定時間が977 μ sのモードとなります。

モジュロ・レジスタに設定する値は、②式より次のようにになります。

$$N = \frac{T}{(分解能)} - 1$$

$$= \frac{120 \times 10^{-6}}{3.8 \times 10^{-6}} - 1$$

$$= 30.5 \doteq 1FH$$

TIMER1:

SEL	MB15	
MOV	XA, #1FH	
MOV	TMOD0, XA	; モジュロ・レジスタを設定
MOV	XA, #01111100B	
MOV	TMO, XA	; タイマ・モード・レジスタを設定

4.2 PTO端子への出力例

μ PD750008は、PTO0またはPTO1端子にタイマ／イベント・カウンタまたはタイマ・カウンタのTOUT F/Fの内容を出力することができます。

TOUT F/Fは、モジュロ・レジスタとカウント・レジスタの値が一致するたびに反転するので、PTO0またはPTO1端子へ方形波を出力することができます。このときの出力周波数は、カウント・パルス周波数をモジュロ・レジスタにしたがって分周した周波数となります。


TOF0の内容をPTO端子に出力するために必要な操作を次に示します。

- ・TO許可フラグをセットする (TOE0←1)
- ・Port2.0の出力ラッチをリセットする (Port2.0←0)
- ・ポート2を出力ポート・モードにする (PM2←1)
- ・ポート2の内蔵プルアップ抵抗の接続を指定しない (POGA.2←0)

4.2.1 メロディ出力

タイマ／イベント・カウンタの出力であるPTO0端子を使用して、外部へメロディを出力するプログラム例を示します。

図4-1 PTO端子の使用例



PTO0端子から出力する周波数は、タイマ／イベント・カウンタ・モード・レジスタによって、決定されるカウント・パルス周波数 (CP) とモジュロ・レジスタに設定される値によって決定されます。

カウント・パルスとして $f_C / 2^4$ ($f_C = 262 \text{ kHz}$: $f_x = 4.194304 \text{ MHz}$ 動作時) を選択したとき、各音階に対してモジュロ・レジスタに設定すべき値およびそれによって得られる周波数の各音階周波数に対する誤差を表4-2に示します。


表4-2 各音階に対してモジュロ・レジスタに設定すべき値、およびそれによって得られる周波数の各音階周波数に対する誤差

平均律音階		PTO0端子出力		誤差 (%)
音階	周波数 (Hz)	TMOD1	周波数 (Hz)	
ド C5	523.25	F9H	523.75	+0.10
ド C5 #	554.37	EBH	554.82	+0.08
レ D5	587.33	DEH	587.16	-0.03
レ D5 #	622.25	D1H	623.51	+0.20
ミ E5	659.26	C6H	657.98	-0.19
ファ F5	698.46	BAH	696.48	-0.28
ファ F5 #	739.99	B0H	739.76	-0.03
ソ G5	783.99	A6H	784.06	+0.01
ソ G5 #	830.61	9CH	828.72	-0.23
ラ A6	880.00	94H	878.78	-0.14
ラ A6 #	932.33	8BH	928.63	-0.40
シ B6	987.77	83H	984.49	-0.33
ド C6	1046.50	7CH	1047.50	+0.10
ド C6 #	1108.73	75H	1109.64	+0.08

メロディは、表4-2に示す各音階の周波数と時間長を変化させて出力します。時間長は、ベーシック・インターバル・タイマを使って作った100 msを基本時間として、その間に何回カウント動作を行うかで変化させます。

図4-2は、P20/PTO0端子からの出力信号の概念図です。

図4-2 P20/PTO0端子からの出力信号の概念図



このプログラム例では、図4-3 データ・フォーマットに従ったデータ・テーブルが、演奏データ・アドレス (MUS) から始まるRAMに格納されており、このデータ・テーブルに従ってメロディを出力します。このデータ・テーブルは、音階を決めるデータと音長を決めるデータの2つで構成されており、この組み合わせを変えることによって、さまざまなメロディを出力することができます。

図4-3 データ・フォーマット

データ	内 容
0XH-FXH	音階データ（表4-4参照）
X0H-XEH	音階データ（表4-5参照）
nFH	$n \times 100\text{ ms}$ の無音を出力後、演奏をはじめから繰り返す

演奏用メモリ・エリアからデータを取り出すことにより、音階を出力できます。

音階出力




表4-3に音階データと音階、表示、およびPTO0端子出力の関係を、表4-4に音長データと音階出力の関係を示します。

表4-3 音階データと音階、表示、および
PTO0端子出力の関係

音階 データ	平均律音階		PTO0端子出力	
	音階	周波数(Hz)	TMOD1	周波数(Hz)
00H	ド C5	523.25	F9H	523.75
01H	ド C5#	554.37	EBH	554.82
02H	レ D5	587.33	DEH	587.16
03H	レ D5#	622.25	D1H	623.51
04H	ミ E5	659.26	C6H	657.98
05H	ファ F5	698.46	BAH	696.48
06H	ファ F5#	739.99	B0H	739.76
07H	ソ G5	783.99	A6H	784.06
08H	ソ G5#	830.61	9CH	828.72
09H	ラ A6	880.00	94H	878.78
0AH	ラ A6#	932.33	8BH	928.63
0BH	シ B6	987.77	83H	984.49
0CH	ド C6	1046.50	7CH	1047.50
0DH	ド C6#	1108.73	75H	1109.64
0EH	無音	0.0	FFH	0.0
0FH	無音	0.0	FFH	0.0

表4-4 音長データと
音階出力の関係

音長 データ	音階出力時間 $T = 100\text{ms}$
00H	$T \times 1$
01H	$T \times 2$
02H	$T \times 3$
03H	$T \times 4$
04H	$T \times 5$
05H	$T \times 6$
06H	$T \times 7$
07H	$T \times 8$
08H	$T \times 9$
09H	$T \times 10$
0AH	$T \times 11$
0BH	$T \times 12$
0CH	$T \times 13$
0DH	$T \times 14$
0EH	$T \times 15$
0FH	リピート

(1) プログラムの説明

このプログラムは、第11章 このアプリケーション・プログラムの応用で使用するものを、入出力インターフェース部を変更して説明しています。

〈使用するレジスタ〉

XA, BC, HL

〈使用するRAM〉

RAMは、メモリ・バンク0の20H番地から配置できます。演奏用メロディ・データはメモリ・バンク1の00H番地からFFH番地に割り当てられます。

MUS :	2ワード	; 演奏データ・アドレス
TONE :	1ワード	; 音階データ
TONETIM :	1ワード	; 音長データ
TONEFLG :	1ワード	; フラグ・エリア
TONE_F :	1ビット EQU	; 1ならば音階出力を許可
TMO_F :	1ビット EQU	; 1ならばタイマ・カウントをスタート

〈ネスティング〉

2レベル（12ワード）

〈使用するハードウェア〉

- ・ポート：Port2.0（PTO0出力兼用端子）
- ・タイマ：ベーシック・インターバル・タイマ、タイマ／イベント・カウンタ

〈初期設定〉


- ・演奏アドレスのクリア
- ・PTO0の出力許可
- ・タイマ／イベント・カウンタの停止


〈起動方法〉


TIM処理を100 msごとに実行することと、M_OUT処理を常に実行することを条件に、MLDYを初期化し、TONE_F = 1 にするとメロディ演奏を開始します。

(2) フロー・チャート

<初期設定>







(3) プログラム例

```

VENTO    MBE=0, RBE=0, INIT      ;RESET
DSEG0   DSEG    0     AT      20H      ;メモリ・バンク0, アドレス20Hから格納
MUS:      DS      2      ;演奏データ・アドレス
TONE:      DS      1      ;音階データ
TONETIM:   DS      1      ;音長データ
TONEFLG:   DS      1      ;フラグ・エリア
TONE_F     EQU     TONEFLG.0  ;1ならば、音階出力を許可
TMO_F     EQU     TONEFLG.1  ;1ならばタイマ・カウント・スタート

; <サブルーチン 初期設定>

CLR1    MBE      ;MBE←0
        MOV     XA, #00H      ;RAMの設定
        MOV     MUS, XA
        MOV     TONE, A
        MOV     TONETIM, A
        MOV     TONEFLG, A
        SET1   TOEO      ;出力許可フラグをセット
        MOV     XA, #00H      ;タイマ／イベント・カウンタを発振停止
        MOV     TMO, XA

```

; <サブルーチン タイマ処理>

```

MOV      A, TONETIM
DECS    A
BR      TIM_116
MOV      XA, MUS

SET1    MBE          ;MBE←1
SEL     MB1          ;メモリ・バンク←1
MOV      HL, XA
MOV      A, @HL+
NOP
MOV      B, A
MOV      A, @HL
MOV      C, A
SEL     MBO          ;メモリ・バンク←0
CLR1    MBE          ;MBE←0

MOV      A, B
MOV      TONETIM, A   ;音長データ←@演奏データ・アドレス
MOV      A, C
MOV      TONE, A       ;音階データ←@演奏データ・アドレス+1
MOV      A, #0H
MOV      T_TIMC, A
SKE     C, #0FH
BR      TIM_115
BR      TIM_114

TIM_115:
SKE     B, #0FH
BR      TIM_111
BR      TIM_112

TIM_112:
MOV      XA, #0H
MOV      MUS, XA
MOV      A, TONE
MOV      TONETIM, A   ;TONETIM←リピート時間
MOV      A, #0FH
MOV      TONE, A       ;TONE←FH

TIM_114:
CLR1    TONE_F
SKE     B, #0FH
BR      TIM_113
RET

TIM_111:
SET1    TONE_F

TIM_113:
MOV      XA, MUS
ADDS    XA, #2H
NOP
MOV      MUS, XA       ;演奏データ・アドレス+2
RET

```

```

TIM_116:
    MOV      TONETIM,A          ;音長データー1
    RET

; <サブルーチン メロディ出力処理>

TONEOUT:
    SKT      TONE_F            ;1ならば音階出力を許可
    BR       TONEO
    MOV      X,#0H
    MOV      A,TONE
    MOV      BC,#HIGH TONE_T
    CALL     !TABLE
    MOV      TMODO,XA           ;モジュロ・レジスタ←音階データ
    MOV      A,TONE
    ADDS   A,#2H
    BR       TONE1

TONE0:
    MOV      XA,#0H             ;タイマ・カウント停止
    CLR1   TM0_F              ;0ならば、タイマ・カウント・スタート
    BR       TONE2

TONE1:
    SKF      TM0_F
    RET
    MOV      XA,#7CH            ;タイマ・カウント・スタート
    SET1   TM0_F              ;1ならば、タイマ・カウント停止

TONE2:
    MOV      TM0,XA
    RET

```

TABLE	CSEG	PAGE
DB	0F9H	;D0
DB	0EBH	;D0#
DB	0DEH	;RE
DB	0D1H	;RE#
DB	0C6H	;MI
DB	0BAH	;FA
DB	0B0H	;FA#
DB	0A6H	;S0
DB	09CH	;S0#
DB	094H	;RA+
DB	08BH	;RA#+
DB	083H	;SI+
DB	07CH	;D0+
DB	075H	;D0#+
DB	0FFH	;無音
DB	0FFH	;無音

```

TABLE:
    MOVT   XA,@BCXA
    RET

```


4.2.2 イベント・パルス分周出力

タイマ／イベント・カウンタのカウント・パルスにTI0端子からの入力パルスを選択し、N分周してP20/PTO0端子へ出力します。ただし、モジュロ・レジスタに設定する値をMとすると

$$N = 2(M+1)$$


となります。

図4-4 イベント・パルス分周出力




入力信号Xと出力信号Yとの関係は次のようにになります。

(a) 4分周の場合 (M = 1, TM0←0CHのとき)



(b) 6分周の場合 (M = 2, TM0←1CHのとき)



次に、TIO端子からの入力パルスを立ち上がりエッジによりカウントし、6分周したパルスをP20/PTO0端子へ出力するプログラムを示します。

このプログラムは、第11章 このアプリケーション・プログラムの応用例には使用されていません。

(1) プログラムの説明

〈外部参照宣言シンボル〉

M

〈使用するレジスタ〉

XA

〈ネスティング〉

1 レベル (4 ワード)

〈使用するハードウェア〉

- ・ポート：Port2.0 (PTO0出力兼用端子)
- ・タイマ：タイマ/イベント・カウンタ

〈初期設定〉

ポート・モード：ポート2←出力モード

〈起動方法〉

タイマ/イベント・カウンタをイベント・カウンタ動作モードに設定し、TOUTの出力を許可し、Port2.0←0とします。

(2) プログラム例

PULSE CSEG INBLOCK

```


CLR1    MBE
MOV     XA, #M
MOV     TMOD0, XA          ; モジュロ・レジスタ設定
MOV     XA, #00001100B
MOV     TMO, XA            ; タイマ・モード・レジスタ設定
SET1    TOE0                ; TOUT出力を許可
CLR1    PORT2.0            ; Port2.0←0
RET

```

4.3 ワンショット・パルス出力


タイマ／イベント・カウンタのPTO端子をワンショット出力に応用します。出力のタイミングはP11/INT1端子のエッジ検出によって行います。

図4-5 ワンショット・パルス出力



このプログラムでは、INT1入力立ち下がりエッジによりPTO0端子へロウ・レベルを出力し、 $(1/16.4 \times 10^3) \times (N_0 + 1)$ 秒後にハイ・レベルを出力します。出力をハイ・レベルに切り替えてから、 $(1/16.4 \times 10^3) \times (N_1 + 1)$ 秒後に再びロウ・レベルを出力します ($N_0 = 1\text{--}255$, $N_1 = 1\text{--}255$)。

図4-6 ワンショット・パルス出力のタイミング (1)




$$t_0 = \frac{1}{16.4} \times (N_0 + 1) \quad (\text{ms})$$

$$t_1 = \frac{1}{16.4} \times (N_1 + 1) \quad (\text{ms})$$

ただし、図4-7のように入力信号の立ち下がりエッジ検出からワンショット出力を終了するまでに、再び立ち下がりエッジを入力しても、このエッジは無視されます。


図4-7 ワンショット・パルス出力のタイミング（2）



(1) 出力時間の性能

入力信号の立ち下がりエッジ検出から、実際にワンショットを出力するまでの時間は、モジュロ・レジスタに設定した値によって決まる時間よりも11マシン・サイクルから13マシン・サイクルの遅れがあります（図4-8参照）。また分解能は、 $1/16.4 \times 10^3 \text{ s} = 6.1 \times 10^{-5} \text{ s}$ となります。

図4-8 ワンショット・パルス出力のタイミング（3）



(2) プログラムの説明

このプログラムは、第11章 このアプリケーション・プログラムの応用では使用されていません。

〈外部参照宣言〉

N0, N1

〈使用するレジスタ〉


XA

〈使用するRAM〉

フラグ用に1ビットのRAMを使用し、メモリ・バンク0の08H-7FH番地に配置することができます。

SHFG : 1ビット ; ディレイ時間が終了し、PTO0端子にハイ・レベルが出力されると
セットされる。

図4-9 プログラム上のRAMのレイアウト



<使用するハードウェア>

- ・ポート：Port2.0（PTO0出力兼用端子），Port1.1（INT1兼用端子）
- ・タイマ：タイマ／イベント・カウンタ

<割り込み>


INT1, INTT0

<初期設定>


- ・RAM : SHFG ← 0
- ・ポート出力 : Port2.0 ← 0
- ・ポート・モード : ポート 2 ← 出力モード
- ・タイマ : TM0 ← 5CH
- ・割り込み : INT1割り込みを許可
- ・INT1 : 立ち下がりエッジ検出モード

(3) フロー・チャート

INT1処理



INTT0処理



(4) プログラム例

```

VENT3      MBE=0, RBE=0, INT1
VENT4      MBE=0, RBE=0, INTT0

DSHOT      DSEG      AT      28H      ;データ・メモリを定義
FLAG:      DS         1H
SHFG       EQU        FLAG.0
PUBLIC     SHFG
EXTRN     N0, N1

; <サブルーチン INT1処理 (MBE = 0) >
INT1      CSEG      INBLOCK
          PUSH      XA
          MOV       XA, #N0
          MOV       TMODO, XA      ;ディレイ時間を設定
          SET1    TM0.3
          CLR1    SHFG
          SET1    TOE0      ;TOUT出力を許可
          DI      IE1       ;INT1を禁止
          EI      IET0      ;INTT0を許可
RETURN:   POP       XA
          RETI

; <サブルーチン INTT0処理 (MBE = 0) >
INTT0:   PUSH      XA
          SKF      SHFG
          BR       OVER

          MOV       XA, #N1
          MOV       TMODO, XA      ;出力時間を設定
          SET1    SHFG
          BR       RETURN

OVER:    CLR1    TOE0      ;出力停止
          DI      IET0      ;TOUT出力を禁止
          EI      IE1       ;INTT0を禁止
          CLR1    IRQ1      ;INT1を許可
          BR       RETURN

```

第5章 時計用タイマの応用

μ PD750008は、時計用タイマを1チャネル内蔵しており、次のような機能があります。

- ① 0.5 sの時間間隔でテスト・フラグ (IRQW) をセットします。またIRQWでスタンバイ・モードを解除することができます。
- ② メイン・システム・クロック、サブシステム・クロックのいずれでも0.5 s間隔を作ることができます。
- ③ 早送りモードにより1/128 (3.91 ms) の時間間隔となり、プログラムのディバグや検査に便利です。
- ④ 固定周波数 (2.048 kHz) をP23/BUZに出力することができ、ブザー音発生や、システム・クロック発振周波数のトリミングに使用できます。
- ⑤ 分周回路のクリアができますから、時計を0秒スタートできます。

5

5.1 時計プログラム

時計用タイマを使った時刻カウント・サブルーチンを紹介します。

このプログラムは、時計用タイマを通常時計モードとし、時計用タイマ割り込み要求フラグ (IRQW) がセットされるごとに秒カウンタをカウントし、120回カウントすると1分とします。

また、分カウンタをカウントするとリターン後スキップします。

(1) プログラムの説明

このプログラムは、第11章 このアプリケーション・プログラムの応用例では使用されていません。

〈パブリック宣言シンボル〉

CNT10：10進カウント・サブルーチン・エントリ・ラベル

CNT 6：6進カウント・サブルーチン・エントリ・ラベル

CNT 7：7進カウント・サブルーチン・エントリ・ラベル

〈外部参照宣言シンボル〉

SECD : 秒カウンタ・アドレス

MIND : 分カウンタ・アドレス

HOURD : 時カウンタ・アドレス

DAYD : 曜日データ・アドレス

〈使用するレジスタ〉


XA, HL

〈使用するRAM〉

アドレス [H]	名 称	用 途	初 期 値
外部参照	SECD (2×4ビット)	秒カウンタ (バイナリ表現で88H-FFHまでカウント)	88H
外部参照	MIND (2×4ビット)	分カウンタ (デシマル表現で0-59までカウント)	0
外部参照	HOURD (2×4ビット)	時カウンタ (デシマル表現で0-23までカウント)	0
外部参照	DAYD (1×4ビット)	曜日データ 日 月 火 水 木 金 土 0 1 2 3 4 5 6	0

RAMアドレス宣言時の条件

- 各シンボルのロウ・アドレスは同じにしてください。また、カラム・アドレスの範囲は0H-EHにしてください。
- 各シンボルは下の図のように配置してください。



〈ネスティング〉

2 レベル (8ワード)

〈使用するハードウェア〉

時計用タイマ（初期設定：通常時計モード）


〈初期設定〉

- 使用するRAMの初期値を設定
- 時計用タイマ・モード設定およびクリア・スタート

〈起動方法〉

- (i) 0.5 s以内に1回以上WATCHをコールします。
- (ii) WATCHをコールするとIRQWの状態をチェックして、次の2つの処理に分かれます。
 - IRQW = 0の場合
RET命令でユーザ・プログラムに戻ります。
 - IRQW = 1の場合
IRQWをリセット(0)したあと時刻のカウントを行い、そのカウント値を秒、分、時間、曜日ごとに各エリアに格納し、RET命令でユーザ・プログラムに戻ります。ただし、分の桁が変化した場合はRETS命令でユーザ・プログラムに戻ります。

(2) フロー・チャート



(3) プログラム例

```

EXTRN DATA(SECD,MIND,HOURD,DAYD)
PUBLIC CNT10,CNT6,CNT7

; <サブルーチン 時計カウント>
WATCH    CSEG      SENT
        CLR1      MBE
        SKTCLR   IROW      ;0.5秒チェック
        RET
        INCS     SECD      ;秒カウンタのカウント
        RET
        INCS     SECD + 1
        RET

        MOV      XA,#100H - 120
        MOV      SECD,XA      ;分カウンタのカウント
        MOV      HL,#MIND
        CALLF   !CNT10
        RETS
        CALLF   !CNT6
        RETS
        CALLF   !CNT10      ;時カウンタのカウント
        BR      WATCH2
        INCS     @HL

WATCH2 :
        INCS     L
        MOV      XA,HOURD
        SKE      A,#4H
        RETS
        SKE      X,#2H
        RETS
        MOV      XA,#00H
        MOV      HOURD,XA
        CALLF   !CNT7      ;曜日を更新
        NOP
        RETS

```

```
; <サブルーチン N進カウント>
CNTN    CSEG      SENT
CNT10:
        MOV       A, #(10H-0AH)      ;10進
CNT6:
        MOV       A, #(10H-6)       ;6進
CNT7:
        MOV       A, #(10H-7)      ;7進
;
        INCS      @HL
        NOP
        ADDS      A, @HL
        BR        NOMRET
        XCH      A, @HL
        INCS      L
        RETS
NOMRET:
        INCS      L
        RET
```


第6章 クロック出力回路の応用

6.1 PCLクロック出力

μ PD750008はクロック出力回路を内蔵しており、P22/PCL端子から周辺LSI、あるいは μ PD7500シリーズなどのスレーブ・マイコンにクロックを供給することができます。

クロック出力の周波数および出力の許可／禁止は、クロック出力モード・レジスタ（CLOM）によって設定します。クロック出力モード・レジスタのフォーマットを図6-1に示します。

図6-1 クロック出力モード・レジスタのフォーマット




注意 CLOMのビット2には、必ず0を書き込むようにしてください。

次にP22/PCL端子にクロックを出力するため必要な操作を示します。

- ・クロック出力モード・レジスタによりクロック出力周波数を設定する。出力は禁止しておく。
- ・Port2.2の出力ラッチをリセットする。
- ・ポート2を出力ポート・モードにする。
- ・クロック出力を許可する。

また、 μ PD750008では、図6-2のようにクロック出力の許可／禁止切り替え時には幅の狭いパルスが出てないようになっています。


図6-2 クロック出力波形



第7章 ビット・シーケンシャル・バッファの応用

ビット・シーケンシャル・バッファ（BSB）はビット操作用の特殊データ・メモリで、Lレジスタによって間接ビット指定ができます。したがって、Lレジスタをインクリメントあるいはデクリメントするだけで、BSB内の指定ビットを更新することができます。

図7-1 ビット・シーケンシャル・バッファの間接アドレシング




次にBSBの応用例を示します。

7.1 高速シリアル・データ転送

μ PD750008で図7-2のようにマルチプロセッサ・システムを構成した場合、ビット・シーケンシャル・バッファを用いてシリアル・データ転送を行います。信号線は1本とし、スレーブCPUからの $\overline{\text{BUSY}}$ 信号出力と送信および受信を兼用します。


図7-2 マルチプロセッサ・システム構成図（高速シリアル・データ転送）



シリアル・データのデータ長はN（1-16）ビットとします。そのフォーマットを次に示します。


・データが0のとき

ロウ・レベルを15マシン・サイクル出力し、続けてハイ・レベルを5マシン・サイクル出力する。




・データが1のとき

ロウ・レベルを5マシン・サイクル出力し、続けてハイ・レベルを15マシン・サイクル出力する。




転送は、スレーブCPUからの $\overline{\text{BUSY}}$ 信号がオン（ハイ・レベル）のときに、マスタCPUからNビットのデータを送信し、続けてスレーブCPUからNビットのデータを送信します。転送データの取り込みは、先頭データの立ち下がりによって開始します。転送の流れを次に示します。


(a) スレーブCPUからの BUSY 信号オン




(b) スレーブCPUからの BUSY 信号オフ



(c) マスタCPUからNビットのデータを出力



(d) スレーブCPUからNビットのデータを出力



(1) プログラムの説明

このプログラムは、第11章 このアプリケーション・プログラムの応用では使用されていません。

〈パブリック宣言シンボル〉

TDATA, RDATA

〈使用するレジスタ〉

XA, L

〈使用するRAM〉

送信データ・エリアと受信データ・エリアを持ち、両エリアの下位アドレスからNビットに転送データを格納します。

RAM上には、メモリ・バンク0の20H-7FH番地に配置することができます。ただし、このときのアドレスは偶数アドレスとします。

TDATA : 4ワード ; 送信データ・エリア。下位アドレスからNビット分のデータを送信する。

RDATA : 4ワード ; 受信データ・エリア。受信したデータを下位Nビットに格納する。

図7-3 プログラム上のRAMのレイアウト

アドレス 40H 41H 42H 43H 44H 45H 46H 47H



〈ネスティング〉

　　レベル1 (4ワード)

〈使用するハードウェア〉

- ・ポート : Port6.0
- ・ビット・シーケンシャル・バッファ 0 - 3

〈初期設定〉


- ・マスタ PCC←3H
Port6.0←入力モード
- ・スレーブ PCC←3H
Port6.0←出力モード
Port6.0←0


〈メイン処理〉

- ・マスタ CALL IBMST
- ・スレーブ CALL IBSLV

(2) マスタCPU

(a) フロー・チャート





(b) プログラム例

```
PUBLIC    TDATA, RDATA  
  
DBS10   DSEG    0      AT      40H  
  
TDATA:  DS      4H      ;送信データ・エリア  
RDATA:  DS      4H      ;受信データ・エリア  
  
N       EQU      8
```

```

BMST    CSEG      INBLOCK
        CLR1      MBE
        SKT       PORT6.0
        RETS

        DI
        SET1      PORT6.0      ;Port6.0←ハイ・レベル
        MOV       XA, #00010000B ;Port6.0を出力モードに設定
        MOV       PMGA, XA
        MOV       XA, TDATA      ;送信データをBSBに設定
        MOV       BSB0, XA
        MOV       XA, TDATA+2
        MOV       BSB2, XA
        MOV       L, #N-1

LOOP:   CLR1      PORT6.0      ;Port6.0←ロウ・レベル
        NOP
        SKF       BSB0. @L    ;Port6.0←BSB0. @L
        SET1      PORT6.0
        MOV       A, #0EH

WAIT:   INC5      A
        BR       WAIT
        NOP
        SKT      PORT6.0
        SET1      PORT6.0      ;Port6.0←ハイ・レベル
        DECS      L
        BR       LOOP

        MOV       XA, #00000000B ;Port6.0を入力モードに設定
        MOV       PMGA, XA

        NOP
        NOP
        NOP

        MOV       L, #N-1


CHECK:  SKF       PORT6.0      ;受信データの立ち下がりをチェック
        BR       CHECK
        NOP
        SET1      BSB0. @L    ;データ取り込み
        SKT      PORT6.0
        CLR1      BSB0. @L
        NOP
        NOP
        NOP
        DECS      L
        BR       CHECK


        MOV       XA, BSB0      ;入力データをRDATAに格納
        MOV       RDATA, XA
        MOV       XA, BSB2
        MOV       RDATA+2, XA
        RET

```

(3) スレーブCPU

(a) フロー・チャート





(b) プログラム例

```
PUBLIC    TDATA, RDATA  
  
DBS10S  DSEG      0      AT      40H  
  
TDATA:  DS        4H          ;送信データ・エリア  
RDATA:  DS        4H          ;受信データ・エリア  
  
N      EQU       8
```

```

BSLV CSEG INBLOCK

DI
CLR1 MBE
MOV L, #N-1
MOV XA, #0000000B ;Port6.0を入力モードに設定
MOV PMGA, XA

CHECK:
SKF PORT6.0 ;受信データの立ち上がりチェック
BR CHECK
NOP
SET1 BSB0. @L ;Port6.0からデータ取り込み
SKT PORT6.0
CLR1 BSB0. @L
NOP
NOP
NOP
DECS L
BR CHECK

NOP
NOP

SET1 PORT6.0 ;Port6.0←ハイ・レベル
MOV XA, #10H ;Port6.0を出力モードに設定
MOV PMGA, XA

MOV XA, BSB0 ;受信データをRDATAに格納
MOV RDATA, XA
MOV XA, BSB2
MOV RDATA+2, XA
MOV XA, TDATA
MOV BSB0, XA
MOV XA, TDATA+2
MOV BSB2, XA
MOV L, #N-1

LOOP:
CLR1 PORT6.0 ;Port6.0←ロウ・レベル
NOP
SKF BSB0. @L ;Port6.0にデータ出力
SET1 PORT6.0
MOV A, #0EH

WAIT:
INCS A
BR WAIT
NOP
SKT PORT6.0
SET1 PORT6.0 ;Port6.0←ハイ・レベル
DECS L
BR LOOP

CLR1 PORT6.0
RET

```

[メモ]


第8章 シリアル・インターフェースの応用

8.1 SBIモードによる応用例

μ PD750008のシリアル・インターフェースをSBIモードで動作させる場合の応用例を示します。

図8-1に、SBIによるシリアル・バスの構成例を示します。

図8-1 シリアル・バスの構成例




(1) マスタCPUの応用

マスタCPUは、スレーブCPUに対して、次の処理を行います。

- (a) アドレスの送信通信を行うスレーブCPUを選択します。
 - (b) コマンドの送信
 - (c) データの送信／受信
-] (a)により選択したスレーブCPUに対して、
コマンドおよびデータにより、通信します。

アドレス、コマンド、データのフォーマットを図8-2に示します。

図8-2 アドレス、コマンド、データのフォーマット




注 データ受信時には、0FFHを書き込む。


また、スレーブCPUから返してくれるアクリング信号（ACK）により、スレーブCPU側で発生したエラーを判断します。

図8-3 アクノリッジ信号によるエラーの判断

・エラー時



・正常時



マスターCPUは、8ビット（1バイト）・データの転送が完了（INTCSIが発生）したあと、スレーブCPUからのアクノリッジ（ACK）信号をチェックします。

転送が完了してから一定の時間内にスレーブCPUよりACK信号が返されなければ、スレーブCPU側でエラーが発生したと判断します。

アドレス、コマンド、データの送信時には、SIOへデータを書き込むとともに、スレーブ・アドレス・レジスタ（SVA）にも同じデータを書き込み、送信データがバス・ライン上で変化した場合のエラー・チェックを行います。

(2) プログラムの説明

このプログラムは、音階や音長の表示データを表示ドライバに転送する例です。

このプログラムは、第11章 このアプリケーション・プログラムの応用で使用するものを、入出力インターフェース部を変更して説明しています。

〈使用するレジスタ〉

レジスタ・バンク0 : XA

レジスタ・バンク3 : XA, BC

〈使用するRAM〉

RAMは、メモリ・バンク0の20H番地から配置します。

MODESBI : 4ワード ; どのデータを転送中かを示す

LCD : 16ワード ; 転送データ・エリア

REFRES_F : 1ビット ; 1ならば転送開始、またはエラー

〈ネスティング〉

1レベル (6ワード)

〈使用するハードウェア〉

ポート : Port0.1 (SCK兼用端子), Port0.2 (SB0兼用端子)

シリアル・インターフェース

〈割り込み〉

INTCSI

〈初期設定〉

- ・ 使用するポートの内蔵プルアップ抵抗の接続を指定

- ・ シリアル動作モード


〈起動方法〉


REFRES_Fをセット(1)後、SBIをスタートすると、LCDの先頭から16ワード分のデータをスレーブCPUのアドレス1H番地に対して転送します。


転送が正常に終了すると、REFRES_Fがリセット(0)されます。


エラーが発生すると、REFRES_Fは1のまま、再度転送を行います。

(3) フロー・チャート









(4) プログラム例

```

VENT4 MBE=0, RBE=1, SBI           ;SBI

DSEG0      DSEG 0      AT      20H      ;メモリ・バンク0, アドレス20Hから格納
MODESBI :   DS      1      ;どのデータを転送中かを示す
LCD :       DS      16     ;LCD表示用エリア

; <サブルーチン 初期設定>
MOV  XA, #0C1H          ;ポート1, 2, 3は内蔵プルアップ抵抗の接続を指定しない
MOV  POGA, XA            ;ポート0, 6, 7は内蔵プルアップ抵抗の接続を指定する
MOV  XA, #8AH            ;シリアル動作モードを設定
MOV  CSIM, XA
SET1 RELT
EI                  ;すべての割り込みを許可

; <サブルーチン SBIスタート>
SKT  REFRES_F           ;1ならばLCDをリフレッシュ
RET
SKT  PORT0.2             ;SB0=ハイ・レベル?
RET
SKT  PORT0.1             ;SCK=ハイ・レベル?
RET
SET1 CMDT                ;CMDTセット
NOP
NOP
SET1 RELT                ;RELTセット
NOP
NOP
SKT  PORT0.2             ;SB0=ハイ・レベル?
RET
SET1 CMDT                ;CMDTセット
MOV  XA, #01H
MOV  S10, XA              ;シフト・レジスタ←スレーブ・アドレス(01H)
MOV  A, #1H
MOV  MODESBI, A           ;MODESBI←1H
EI   IECSI                ;SBI割り込みを許可
MOV  XA, #1H
MOV  SVA, XA              ;スレーブ・アドレス・レジスタ←1H
CLR1 REFRES_F             ;REFRES_F←0
RET
;

```

```

; <サブルーチン SBI処理>

SBI:
    SEL RB3           ;レジスタ・バンク←3
    DI IEBT          ;ベーシック・インターバル・タイマ(INTBT)の使用
                      ;を禁止
    EI               ;すべての割り込みを許可
    MOV A, MODESB1
    ADDS A, #0FH
    BR SB1_E
    ADDS A, #0FH
    BR SB11
    ADDS A, #0FH
    BR SB12
    ADDS A, #0FH
    BR SB13
    ADDS A, #0FH
    BR SB14
    ADDS A, #0FH
    BR SB15
    ADDS A, #0FH
    BR SB16
    ADDS A, #0FH
    BR SB17
    ADDS A, #0FH
    BR SB18
    ADDS A, #0FH
    BR SB19
    ADDS A, #0FH
    BR SB110
    ADDS A, #0FH
    BR SB111
    BR SB1_E

SB11:
    MOV BC, #04H      ;シフト・レジスタ←ポインタ・コマンド(04H)
    BR SB1TIM

SB12:
    MOV BC, #07H      ;シフト・レジスタ←ポインタ(07H)
    BR SB1TIM

SB13:
    MOV BC, #01H      ;シフト・レジスタ←ライト・コマンド(01H)
    BR SB1TIM

SB14:
    MOV XA, LCD       ;シフト・レジスタ←アドレス1桁目
    MOV BC, XA
    BR SB1TIM

SB15:
    MOV XA, LCD+2     ;シフト・レジスタ←アドレス2桁目
    MOV BC, XA
    BR SB1TIM

```

```

SB16:
    MOV    XA,LCD+4          ;シフト・レジスタ←アドレス3桁目
    MOV    BC,XA
    BR     SBITIM

SB17:
    MOV    XA,LCD+6          ;シフト・レジスタ←“_”データ
    MOV    BC,XA
    BR     SBITIM

SB18:
    MOV    XA,LCD+8          ;シフト・レジスタ←音階データ1桁目
    MOV    BC,XA
    BR     SBITIM

SB19:
    MOV    XA,LCD+10         ;シフト・レジスタ←音階データ2桁目
    MOV    BC,XA
    BR     SBITIM

SB110:
    MOV    XA,LCD+12         ;シフト・レジスタ←音階データ3桁目
    MOV    BC,XA
    BR     SBITIM

SB111:
    MOV    XA,LCD+14         ;シフト・レジスタ←音階データ4桁目
    MOV    BC,XA
SBITIM:
    MOV    L,#1110B           ;100 μsカウンタの初期化
SBICNT:
    DECS   L
    BR     SBI_ACKD

; <サブルーチン エラー処理>

SB10:
    MOV    A,#0H
    MOV    MODESBI,A          ;MODESBI←0
    SET1   REFRES_F           ;1ならばLCD表示をリフレッシュ
    BR     SBI_E

SBI_ACKD:
    SKT    ACKD               ;ACKノリッジ検出フラグ=1 ?
    BR     SBICNT
    SKT    COI                ;アドレス・コンパレータ信号=1 ?
    BR     SB10
    SKT    PORTO.2             ;SB0=ハイ・レベル ?
    BR     SBI_E
    SKT    PORTO.1             ;SCK=ハイ・レベル ?
    BR     SBI_E
    MOV    A,MODESBI
    ADDS  A,#0FH               ;MODESBI=0 ?
    BR     SBI_S10
    ADDS  A,#0FH               ;MODESBI=1 ?
    BR     SBI_CMDT

```


```
ADD$ A, #0FH ;MODESB1=2 ?
BR SBI_S10
ADD$ A, #0FH ;MODESB1=3 ?
SBI_CMDT:
SET1 CMDT ;CMDTセット
SBI_S10:
MOV XA, BC
MOV S10, XA ;シフト・レジスタ←BCレジスタ
INCS MODESB1 ;MODESB1+1
NOP
SET1 CSIE ;シリアル・インターフェース動作、シフト・レジスタの
           ;使用を禁止
EI IECSI ;SBI割り込みを許可
SBI_E:
RETI ;
;
```

第9章 キー入力サブルーチン

メンタリ・キー (3×8) のキー入力を行います。

ここでは、キー・スキャン信号出力にポート3、リターン信号の取り込みにポート6、7を使用して、図9-1のようなキー・マトリクスを構成します。

図9-1 キー・マトリクスの構成



このプログラムでは、8 msごとにキー・スキャンを行い、3回連続でキー・データが一致するとキーが確定 (KEYCHKF = 1) します。

入力したキー・データは4ビットのキー・ソース・データ (KEYS) と8ビットのキー・リターン・データ (KEYR) に格納します。各キーとキー・データの関係を次に示します。

各キーとキー・データの関係

キー	KEYS KEYR
K00	: 1110B 11111110B
K01	: 1110B 11111101B
K02	: 1110B 11111011B
K03	: 1110B 11110111B
K04	: 1110B 11101111B
K05	: 1110B 11011111B
K06	: 1110B 10111111B
K07	: 1110B 01111111B
K08	: 1101B 11111110B
K09	: 1101B 11111101B
K10	: 1101B 11111011B
K11	: 1101B 11110111B
K12	: 1101B 11101111B
K13	: 1101B 11011111B
K14	: 1101B 10111111B
K15	: 1101B 01111111B
K16	: 1011B 11111110B
K17	: 1011B 11111101B
K18	: 1011B 11111011B
K19	: 1011B 11110111B
K20	: 1011B 11101111B
K21	: 1011B 11011111B
K22	: 1011B 10111111B
K23	: 1011B 01111111B

(1) プログラムの説明

このプログラムは、**第11章 このアプリケーション・プログラムの応用**で使用するものを、入出力インターフェース部を変更して説明しています。

〈使用するレジスタ〉

XA, BC, HL

〈使用するRAM〉

RAMは、メモリ・バンク0の20H番地から配置します。

LEDDIG :	1ワード	; LED表示桁番号。2msごとに次のように変化する ; 0111B→1011B→1101B→1110B
KEYCODE :	2ワード	; キー・ソース、キー・リターンをコード化する
KEYR :	2ワード	; キー・リターン
KEYRB :	2ワード	; 前回のキー・リターン
KEYTIM :	2ワード	; キー・オン時間カウント用タイマ
KEYS :	1ワード	; キー・ソース
KEYSB :	1ワード	; 前回のキー・ソース
KEYCATT :	1ワード	; キー・チャタリング・カウンタ
KEYF :	1ワード	; フラグ
KEYONF :	1ビット	; 1ならばキー・オン
KEYON2SF :	1ビット	; 2秒以上キーを押し続けると、1になる
HTKEYONF :	1ビット	; 1ならば本体のキー・オン
KEYCHKF :	1ビット	; 1ならばキー・スキャンを終了

〈ネスティング〉

1レベル（6ワード）

〈使用するハードウェア〉

ポート：ポート3, 6, 7

〈初期設定〉

- ポート3を出力モードにする。
- ポート6, 7を入力モードにして、内蔵プルアップ抵抗の接続を指定する。


〈起動方法〉

ベーシック・インターバル・タイマ割り込みは2msごとに発生するとします。

キー・チェック用の8msのタイマはLED表示用の桁カウンタ（LEDDIG）を利用します。

メイン・ルーチンのキー・デコード処理でキー・スキャンの終了判定を行い、各キーを処理します。

(2) フロー・チャート



(3) プログラム例

```

DSEG0      DSEG 0    AT     20H      ;メモリ・バンク0, アドレス20Hから格納
LEDDIG:     DS      1          ;LED表示桁番号
KEYCODE:    DS      2          ;キー・ソース, キー・リターンをコード化
KEYR:       DS      2          ;キー・リターン
KEYRB:      DS      2          ;前回のキー・リターン
KEYTIM:    DS      2          ;キー・オン時間カウント用タイマ
KEYS:       DS      1          ;キー・ソース
KEYSB:      DS      1          ;前回のキー・ソース
KEYCATT:   DS      1          ;キーのチャタリング・カウンタ

KEYFLG:    DS      1          ;フラグ
KEYONF:    EQU    KEYFLG.3   ;1ならばキー・オン
KEYON2SF:  EQU    KEYFLG.2   ;1ならば2秒以上キーを押し続けている
HTKEYONF:  EQU    KEYFLG.1   ;1ならば本体キー・オン
KEYCHKF:  EQU    KEYFLG.0   ;1ならばキー・スキャン終了

;***** KEYRデータ *****

KEYR_1     EQU    11111110B
KEYR_2     EQU    11111101B
KEYR_3     EQU    11111011B
KEYR_4     EQU    11110111B
KEYR_5     EQU    11101111B
KEYR_6     EQU    11011111B
KEYR_7     EQU    10111111B
KEYR_8     EQU    01111111B

; <サブルーチン 初期化>
MOV A, #0FH
OUT PORT3, A
MOV XA, #0FH
MOV PMGA, XA           ;ポート3を出力モードに, ポート6を入力モードに
MOV XA, #34H
MOV PMGB, XA           ;ポート2, 4, 5を出力モードに, ポート7を入力モードに
MOV XA, #0C1H           ;ポート1, 2, 3に内蔵プルアップ抵抗を接続
MOV POGA, XA           ;ポート0, 6, 7に内蔵プルアップ抵抗を接続

```

```

; メイン・ルーチン
;+++++
; キー・デコード
;+++++
KEYDEC:
    SKT    KEYCHKF      ; 1ならばキー・スキャン終了
    BR     KEYEND

    CLR1   KEYCHKF
    SKT    KEYONF       ; 1ならばキー・オン
    BR     KEYOFF

;-----キー・オン処理-----
;-----BR     KEYEND
;

KEYOFF:
;-----キー・オフ処理-----
;

KEYEND:
;-----<サブルーチン 100 ms タイマ>
;-----キー・オン時間のカウント-----
    MOV    XA KEYTIM
    MOV    HL, XA
    INC$   HL
    NOP
    MOV    XA, HL
    MOV    KEYTIM, XA      ; キー・オン時間カウント用タイマ+1
;
```

```

; <サブルーチン 2 msのベーシック・インターバル・タイマ割り込み>
;+++++
; キー・チェック
;+++++
KEYCHK:
    MOV     A,LEDDIG
    SKE     A,#0111B           ;8 ms?
    RET

    MOV     A,#0H
    OUT    PORT3,A            ;ポート3←0
    MOV     XA,#0FFH
    OUT    PORT4,XA            ;ポート4, 5←ALL1
    IN     XA,PORT6
    MOV     BC,#0FFH
    SKE     XA,BC              ;ポート6, 7=ALL1?
    BR     KEYCHK1
    SKT     KEYONF             ;1ならばキー・オン
    RET
    INCs   KEYCATT             ;チャタリング・カウンタ+1
    NOP
    MOV     A,KEYCATT
    ADDS   A,#0DH               ;キーが3回一致?
    RET
    MOV     A,#0H
    MOV     KEYCATT,A           ;チャタリング・カウンタ←0H
    CLR1   KEYONF              ;KEYONF←0
    SET1   KEYCHKF              ;1ならばキー・スキャンを終了
    RET

KEYCHK1:
    MOV     A,#1110B
    MOV     KEYS,A               ;キー・ソース←1110B
KEYCHK2:
    OUT    PORT3,A            ;ポート3←1110B
    NOP
    NOP
    NOP
    NOP
    IN     XA,PORT6
    MOV     KEYR,XA              ;キー・リターン←ポート6, 7
    MOV     BC,#0FFH
    SKE     XA,BC              ;ポート6, 7=ALL1?
    BR     KEYCHK3
    SET1   CY                  ;CY←1
    MOV     A,KEYS
    ADDC   XA,XA
    MOV     KEYS,A               ;キー・ソースを左へ1ビット・シフト
    SKE     A,#0111B
    BR     KEYCHK2
    RET

```

KEYCHK3:

```

MOV    A, KEYS
MOV    HL, #KEYSB
SKE    A, @HL           ;キー・ソース=KEYSB?
BR    KEYCHK23
MOV    A, KEYR
MOV    HL, #KEYRB
SKE    A, @HL           ;キー・リターン=KEYRB?
BR    KEYCHK23
MOV    A, KEYR+1
MOV    HL, #KEYRB+1
SKE    A, @HL           ;キー・リターン+1=KEYRB+1?
BR    KEYCHK23
INCS   KEYCATT          ;チャタリング・カウンタ+1
NOP
MOV    A, KEYCATT
ADDS   A, #0DH           ;キーが3回一致?
RET
;
;
```

; <サブルーチン チャタリング除去終了>

```

SET1   KEYONF
CLR1   KEYON2SF
MOV    XA, #0H
MOV    KEYTIM, XA         ;2sタイマを起動
SET1   KEYCHKF          ;1ならばキー・スキャンを終了
RET
;
```

; <サブルーチン 前回とキーの内容が違う>

KEYCHK23:

```


MOV    A, KEYS
MOV    KEYSB, A           ;KEYSB←キー・ソース
MOV    XA, KEYR
MOV    KEYRB, XA          ;KEYRB←キー・リターン
MOV    A, #0H
MOV    KEYCATT, A          ;チャタリング・カウンタ←0
RET
;
```

第10章 各種サブルーチン

10.1 データ転送

メモリ・バンク0とMBE・MBSによって選択されたメモリ・バンクとの間で最大16ワードまでのデータの交換を行います。

交換するデータの先頭アドレスをDEレジスタ（メモリ・バンク = 0）とHLレジスタ（メモリ・バンク = MBE・MBS）で指定し、データ長（ワード数）をBレジスタで指定します。ただし、HLレジスタ、DEレジスタで指定するアドレスは、それぞれ2行にわたらないものとします。



10

プログラム例

このプログラムは、第11章 このアプリケーション・プログラムの応用例には使用されていません。

- ・ステップ数：9ステップ（9バイト）
- ・使用するレジスタ：A, E, DE, HL


```
EXCH      CSEG      INBLOCK
          EXCH:
          XCH      A, @DE
          XCH      A, @HL+
          XCH      A, @DE
          INCSD    E
          NOP
          DECS    B
          BR      EXCH
          RET
```

10.2 データ比較

メモリ・バンク0とMBE・MBSによって選択されたメモリ・バンクとの間で最大16ワードまでのデータの比較を行います。

比較するデータの先頭アドレスをDEレジスタ（メモリ・バンク = 0）とHLレジスタ（メモリ・バンク = MBE・MBS）で指定し、データ長（ワード数）をBレジスタで指定します。ただし、HLレジスタ、DEレジスタで指定するアドレスは、それぞれ2行にわたらないものとします。

比較した結果、2つのデータが等しいときはリターン・スキップします。



プログラム例

このプログラムは、第11章 このアプリケーション・プログラムの応用例には使用されていません。

- ・ステップ数：10ステップ（10バイト）
- ・使用するレジスタ：A, B, DE, HL

```

COMP    CSEG    INBLOCK

    MOV     A, @DE
    SKE     A, @HL
    RET
    INC$    L
    NOP
    INC$    E
    NOP
    DEC$    B
    BR     COMP

RETS


```

10.3 10進加算

メモリ・バンク0とMBE・MBSによって選択されたメモリ・バンクとの間で、最大16桁（16ワード）までの10進加算を行います。

10進加算するデータの最下位桁アドレスをDLレジスタとHLレジスタで指定し、桁数は10H-Lレジスタの値とします。このとき@DLで示されるデータ・メモリのアドレスは、MBE・MBSの値にかかわらず000H-0FFH（メモリ・バンク0）であり、@HLで示されるデータ・メモリのアドレスは、MBE・MBSの値で示されるメモリ・バンクの範囲となります。

10進加算した結果は、DLレジスタでアドレッシングしたデータ・エリアに格納します。また、オーバフローはキャリー・フラグ（CY）に残します。



プログラム例

このプログラムは、第11章 このアプリケーション・プログラムの応用例には使用されていません。

- ・ステップ数：9ステップ（9バイト）
- ・使用するレジスタ：A, D, HL

```

DECADD CSEG INBLOCK
    CLR1 CY
LOOP:
    MOV A, @DL
    ADDS A, #6
    ADDC A, @HL
    ADDS A, #0AH
    XCH A, @DL
    INC S L
    BR LOOP
    RET


```

10.4 10進減算

メモリ・バンク0とMBE・MBSによって選択されたメモリ・バンクとの間で、最大16桁（16ワード）までの10進減算を行います。

10進減算の被減数を格納するデータ・エリアの最下位桁アドレスをDLレジスタで指定し、減数を格納するデータ・エリアの最下位桁アドレスをHLレジスタで指定します。桁数は10H-Lレジスタの値とします。このとき@DLで示されるデータ・メモリのアドレスは、MBE・MBSの値にかかわらず000H-0FFH（メモリ・バンク0）であり、@HLで示されるデータ・メモリのアドレスは、MBE・MBSの値で示されるメモリ・バンクの範囲となります。

10進減算した結果は、HLレジスタでアドレシングしたデータ・エリアに格納します。また、ボローはキャリー・フラグ（CY）に残します。



プログラム例

このプログラムは、第11章 このアプリケーション・プログラムの応用例には使用されていません。

- ・ステップ数：19ステップ（19バイト）
- ・使用するレジスタ：A, D, HL
- ・使用するスタック：2ワード

```

DECSUB CSEG    INBLOCK

        PUSH   HL
        CLR1   CY
LOOP0:
        MOV    A,@DL
        SUBC  A,@HL
        ADDS  A,#0AH
        XCH   A,@HL
        INCs  L
        BR    LOOP0

        POP    HL
        SKT   CY
        RET


        CLR1   CY
LOOP1:
        MOV    A,#0H
        SUBC  A,@HL
        ADDS  A,#0AH
        XCH   A,@HL
        INCs  L
        BR    LOOP1

        RET

```

10.5 2進10進変換

データ・メモリ内の2進8ビット・データを2進化10進数に変換し、変換結果を変換前にデータがあったデータ・メモリとその上位1ワードに格納します。



2進8ビット・データは、下位桁をX、上位桁をYとすると

$$X_{(2\text{進})} + Y_{(2\text{進})} \times 10 + 6$$

となり、


$$(X_{(2\text{進})} + Y_{(2\text{進})} \times 10) + Y_{(2\text{進})} \times 6$$

と表すことができます。


ここで、XとYは2進数なので、まず $(X + Y \times 10)$ の部分のXとYを10進変換し、次にYの値の回数だけ6を加算します。このとき、加算するたびに10進補正を行います。

以上の処理によって、もとのデータを10進変換した値を得ることができます。

(1) 使用するRAM



(2) フロー・チャート



(3) プログラム例

このプログラムは、第11章 このアプリケーション・プログラムの応用例には使用されていません。

- ・ステップ数：24ステップ（32バイト）

- ・使用するレジスタ：A, E, HL

```

PUBLIC WORK

BDDATA DSEG AT 10H

WORK: DS 3H
BDCNV CSEG INBLOCK

MOV A,#0H
MOV WORK + 2,A
MOV A,WORK + 1
XCH A,E

MOV A,WORK

BD:
MOV HL,#WORK ;10進補正(1桁目)
ADDS A,#6H
BR ADD10

MOV @HL,A ;2桁目をインクリメント
INCS L
MOV A,#7H
ADDS A,@HL ;10進補正
BR ADD10

INCS WORK + 2 ;3桁目をインクリメント

CTLLOOP:
MOV @HL,A ;2桁目の値の回数、6を加算
DECS E
BR BDX2
RET

BDX2:
MOV A,WORK
ADDS A,#6H
BR BD

ADD10:
ADDS A,#0AH
NOP
BR CTLLOOP


```

10.6 8ビット乗算

データ・メモリ内の8ビット・ワーク・エリアWORKAを被乗数とし、8ビット・ワーク・エリアWORKCを乗数として、8ビットの2進乗算を行います。

演算結果は、上位8ビットをWORKAに格納し、下位8ビットを8ビット・ワーク・エリアWORKBに格納します。

ただし、WORKA, WORKBは、下位アドレスから連続し、そのメモリ・バンクは、MBE・MBSによって選択されたメモリ・バンクとします。また、WORKCのメモリ・バンクは、0とします。



プログラム例

このプログラムは、第11章 このアプリケーション・プログラムの応用例には使用されていません。

- ・ステップ数：38ステップ（50バイト）
- ・使用するレジスタ：XA, BC, DE, HL
- ・使用するスタック：2ワード

```
PUBLIC WORKA, WORKB, WORKC
```

```
DMLT8 DSEG AT 20H
```

```
WORKC: DS 2H
WORKB: DS 2H
WORKA: DS 2H
```

```
MULT8 CSEG INBLOCK
```

```
MOV XA, #00H
MOV WORKB, XA
MOV C, #1H
```

```
LOOP:
MOV B, #3H
MOV HL, #WORKB
MOV A, #0H
```

SHLOOP:

```

XCH    A, @HL+
NOP
DECS   B
BR     SHLOOP

```

```
XCH    A, B
```

FIGURE:

```

DECS   B
BR     ADD

```

```

DECS   C
BR     LOOP

```

```
RET
```

ADD:

```

PUSH   BC
CLR1   CY
MOV    B, #1H
MOV    DE, #WORKC
MOV    HL, #WORKB

```

ADLOOP:

```

MOV    A, @DE
ADDC   A, @HL
XCH    A, @HL+
INCS   L
NOP
INCS   E
NOP
DECS   B
BR     ADLOOP

```

```

POP    BC
MOV    HL, #WORKA
NOT1   CY
SKT    CY

```

INCA:

```

INCS   @HL
BR     FIGURE

```

```

INCS   L
NOP
BR     INCA


```

10.7 16ビット乗算

データ・メモリ内の16ビット・ワーク・エリアWORKAを被乗数とし、16ビット・ワーク・エリアWORKCを乗数として、16ビットの2進乗数を行います。

演算結果は、上位16ビットをWORKBに格納し、下位16ビットを16ビット・ワーク・エリアWORKAに格納します。

ただし、WORKA, WORKBは、下位アドレスから連続し、そのメモリ・バンクは、MBE・MBSによって選択されたメモリ・バンクとします。また、WORKCのメモリ・バンクは、0とします。



プログラム例

このプログラムは、第11章 このアプリケーション・プログラムの応用例には使用されていません。

- ・ステップ数：39ステップ（51バイト）
- ・使用するレジスタ：XA, BC, DE, HL
- ・使用するスタック：2ワード

```

PUBLIC WORKA, WORKB, WORKC

DMLT16 DSEG AT 20H

WORKC: DS 4H
WORKB: DS 4H
WORKA: DS 4H

MULT16 CSEG INBLOCK

    MOV XA, #00H
    MOV WORKB, XA
    MOV WORKB+2, XA
    MOV C, #3H
LOOP:
    MOV B, #7H
    MOV HL, #WORKB
    MOV A, #0H
SHLOOP:
    XCH A, @HL+
    NOP
    DECS B
    BR SHLOOP

    XCH A, B
FIGURE:
    DECS B
    BR ADD
    DECS C
    BR LOOP
    RET
ADD:
    PUSH BC
    CLR1 CY
    MOV B, #3H
    MOV DE, #WORKC
    MOV HL, #WORKB
ADLOOP:
    MOV A, @DE
    ADDC A, @HL
    XCH A, @HL+
    NOP
    INCs E
    NOP
    DECS B
    BR ADLOOP

    POP BC
    MOV HL, #WORKA
    NOT1 CY
    SKT CY
INCA:
    INCs @HL
    BR FIGURE

    INCs L
    NOP
    BR INCA



```

10.8 2進除算（16ビット÷8ビット）

データ・メモリ内の8ビット・ワーク・エリアWORKBおよびWORKCを被除数とし、8ビット・ワーク・エリアWORKDを除数として、2進除算を行います。

演算結果は、上位8ビットをWORKCに格納し、下位8ビットをWORKBに格納します。また、8ビット・ワーク・エリアWORKAに余りを格納します。

ただし、WORKA, WORKB, WORKCは同一のロウ・アドレス上にあり、下位アドレスから連続しているものとします。またWORKDのCOLUMNアドレスはWORKAと同じものとし、いずれのワーク・エリアもメモリ・バンク0とします。



プログラム例

このプログラムは、第11章 このアプリケーション・プログラムの応用例には使用されていません。

- ・ステップ数：42ステップ（59バイト）
- ・使用するレジスタ：XA, B, D, HL

```

PUBLIC WORKA, WORKB, WORKC, WORKD

DIV8D0 DSEG AT 20H
WORKA: DS 2H
WORKB: DS 2H
WORKC: DS 2H

DIV8D1 DSEG AT 30H
WORKD: DS 2H

DIV8 CSEG INBLOCK
CLR1 MBE
MOV B, #0H
MOV XA, #00H
MOV WORKA, XA
LOOP:
MOV HL, #WORKA
SET1 CY
SKT (WORKA + 5).3
CLR1 CY
ROTAT:
MOV A, @HL
ADDC A, @HL
MOV @HL, A
INCS L
NOP
SKE L, #(WORKA + 6) AND 0FH
BR ROTAT

CLR1 CY
MOV HL, #WORKD
MOV D, #WORKA SHR 4

```

SUB:

```
MOV    A, @DL
SUBC   A, @HL
XCH    A, @DL
INCS   L
SKE    L, #(WORKA + 2) AND OFH
BR     SUB

SKT    CY
BR     SETB0

SKT    WORKB.0
BR     ADD
```

FIGCNT:

```
INCS   B
BR     LOOP
RET
```

ADD:

```
CLR1   CY
MOV    L, #WORKA AND OFH
```

ADDLP:

```
MOV    A, @DL
ADDC   A, @HL
XCH    A, @DL
INCS   L
SKE    L, #(WORKA + 2) AND OFH
BR     ADDLP
BR     FIGCNT
```

SETB0:

```
SET1   WORKB.0
BR     FIGCNT
```


10.9 2進除算 (16ビット÷16ビット)

データ・メモリ内の16ビット・ワーク・エリアWORKAを被除数とし、16ビット・ワーク・エリアWORKCを除数として、2進除算を行います。

演算結果は、WORKAに商を格納し、16ビット・ワーク・エリアWORKBに余りを格納します。

ただし、WORKA, WORKBは、下位アドレスから連続しているものとし、メモリ・バンクは0とします。

また、WORKCのメモリ・バンクは、MBE・MBSによって選択されたメモリ・バンクとします。



プログラム例

このプログラムは、第11章 このアプリケーション・プログラムの応用例には使用されていません。

- ・ステップ数：49ステップ（63バイト）
- ・使用するレジスタ：A, B, DE, HL

```

PUBLIC WORKA, WORKB, WORKC

DDIV16 DSEG AT 20H

WORKA: DS 4H
WORKB: DS 4H
WORKC: DS 4H

DIV16 CSEG INBLOCK

        MOV DE, #WORKB
CLRB:   MOV A, #0H
        XCH A, @DE
        INCs E
        NOP

        SKE E #(WORKB + 4) AND 0FH
        BR CLRBL

        MOV B, #0CH
LOOP:   MOV DE, #WORKA
        MOV A, #0H
SHFT:   XCH A, @DE
        INCs E
        NOP
        SKE E, #(WORKA + 8) AND 0FH
        BR SHFTL

SUB:    MOV DE, #WORKB
        MOV HL, #WORKC
        CLR1 CY

SLOOP:  MOV A, @DE
        SUBC A, @HL
        XCH A, @DE
        INCs L
        NOP
        INCs E
        NOP
        SKE L, #(WORKA + 4) AND 0FH
        BR SLLOP

```

```
NOT1    CY
SKT     CY
BR      DO

MOV     DE, #WORKA
XCH     A, @DE
ADDS   A, #1H
XCH     A, @DE
BR      SUB
DO:
MOV     DE, #WORKB
MOV     HL, #WORKC
ADD:
MOV     A, @DE
ADDC   A, @HL
XCH     A, @DE
INCS   L
NOP
INCS   E
NOP
SKE    E, #(WORKB + 4) AND 0FH
BR     ADD
INCS   B
BR     LOOP

RET
```

第11章 このアプリケーション・プログラムの応用例

この章では、前章までのアプリケーション・プログラムを使用して作成したシステム・プログラムの例を示します。前章までのアプリケーション・プログラムのうち、使用したもの次に示します。

3.3 リモコン信号受信応用

4.2.1 メロディ出力


8.1 SBIモードによる応用例

第9章 キー入力サブルーチン

11.1 プログラム概要

このプログラムは、鍵盤に見立てたモメンタリ・キーまたはリモコン・キーにより、音階データとキーを押している時間による音長データを入力し、自動的に演奏を行います。

次に、全体のフロー・チャートを示します。



注1. 4.2.1 メロディ出力のアプリケーション・ソフトを使用。

2. 3.3 リモコン信号受信応用のアプリケーション・ソフトを使用。

3. 第9章 キー入力サブルーチンのアプリケーション・ソフトを使用。


4. 8.1 SBIモードによる応用例のアプリケーション・ソフトを使用。

11.2 システム構成

このシステムは、デバイスとしてμPD750008を使用し、キー・マトリクスまたはリモコン入力により、14セグメントLCDドライバ、7セグメントLEDドライバ、および圧電ブザーへの出力を制御します。

図11-1にシステム構成図を示します。

図11-1 システム構成図



- キー・マトリクス : MODE, UP, DOWN, END, WAIT, 音階の各キーが配置されています。
- 赤外線受光部 : NECフォーマットの赤外線コードを受光します。
MODE, UP, DOWN, END, WAIT, 音階の各キーがコード化されています。
- 7セグメントLED : 4桁の7セグメントLEDに“PROG”, “PLAY”, “PAUS” のモード, および
100 ms-900 msまでの音長を表示します。ディジット信号は, キー・マトリクス
のキー・ソース信号と共にします。
- LCD ドライバ : SBIフォーマットのシリアル・データにより, 8桁の14セグメントLCDに演奏
データの内容, または出力中の音階を表示します。
- 圧電ブザー : 押されたキー, または演奏データを出力します。

11.3 ポート割り付け

表11-1 μ PD750008のポート割り付け (1/2)

端子番号	端子名	入出力	兼用端子	機能	リセット時	アクティブ値	初期設定値
15	P00	入力	INT4	未使用 (GNDに接続)	入力	—	入
14	P01	入出力	SCK	シリアル・シフト・クロック		L	H
13	P02	入出力	SO/SBO	LCD ドライバとのシリアル・バス・ライン		—	H
12	P03	入出力	SI/SB1	未使用 (GNDに接続)		—	入
19	P10	入力	INT0	リモコン入力	入力	L	入
18	P11		INT1	未使用 (GNDに接続)		—	入
17	P12		INT2	未使用 (GNDに接続)		—	入
16	P13		TI0	未使用 (GNDに接続)		—	入
25	P20	入出力	PTO0	圧電ブザー	入力	L	H
24	P21		PTO1	未使用 (GNDに接続)		—	入
23	P22		PCL	未使用 (GNDに接続)		—	入
22	P23		BUZ	未使用 (GNDに接続)		—	入
9	P30	入出力	—	ディジット出力 (LED7) /キー・ストローブ信号	入力	L	H
8	P31		—	ディジット出力 (LED6) /キー・ストローブ信号		L	H
7	P32		—	ディジット出力 (LED5) /キー・ストローブ信号		L	H
6	P33		—	ディジット出力 (LED4) /キー・ストローブ信号		L	H
41	P40	入出力	—	セグメント出力 (D.P.)	ハイ・レベル (ブルアップ抵抗内蔵時) またはハイ・インピーダンス	L	H
40	P41		—	セグメント出力 (g)		L	H
39	P42		—	セグメント出力 (f)		L	H
38	P43		—	セグメント出力 (e)		L	H
37	P50	入出力	—	セグメント出力 (d)	ハイ・レベル (ブルアップ抵抗内蔵時) またはハイ・インピーダンス	L	H
36	P51		—	セグメント出力 (c)		L	H
35	P52		—	セグメント出力 (b)		L	H
34	P53		—	セグメント出力 (a)		L	H
33	P60	入出力	KR0	キー入力端子	入力	L	H
32	P61		KR1	キー入力端子		L	H
31	P62		KR2	キー入力端子		L	H
30	P63		KR3	キー入力端子		L	H
29	P70	入出力	KR4	キー入力端子	入力	L	H
28	P71		KR5	キー入力端子		L	H
27	P72		KR6	キー入力端子		L	H
26	P73		KR7	キー入力端子		L	H
11	P80	入出力	—	未使用 (GNDに接続)	入力	—	入
10	P81		—	未使用 (GNDに接続)		—	入

表11-1 μ PD750008のポート割り付け (2/2)

端子番号	端子名	入出力	兼用端子	機能	リセット時	アクティブ値	初期設定値
4	X1, X2	入力	—	メイン・システム・クロック発振用 クリスタル／セラミック接続端子	入力	—	—
5						—	—
1	XT1	入力	—	未使用 (GNDに接続)	入力	—	入
2	XT2	—		未使用 (GNDに接続)	—	—	入
3	RESET	入力	—	システム・リセット入力端子	—	L	—
20	IC	—	—	未使用 (GNDに接続)	—	—	入
21	V _{DD}	—	—	未使用 (GNDに接続)	—	—	入
42	V _{SS}	—	—	未使用 (GNDに接続)	—	—	入

11.4 状態遷移表

表11-2 状態遷移表

キー入力 モード	モード・キー	音階キー	ENDキー	UPキー	DOWNキー
P R O G 機能	PAUSモードに切り替える。	キー・オンが2秒未満 キー・オンが2秒以上	キー・オン キー・オフ	キー・オン キー・オフ	演奏アドレスを+1 演奏アドレスを-1
P L A Y 機能	PROGモードに切り替える。データの入力は演奏アドレスの続きから行う。	データの入力を、先頭のデータから行うか、終わりのデータから行うか切り替える。	無音を出力し、音長カウントを停止する。また、音長をカウントする。	音長カウンタをカウントする。音階を演奏アドレスに記憶し、次のアドレスへ進む。	演奏アドレスを停止し、リピートコードと音長データ(リピート時間)を演奏アドレスに記憶する。アドレスは変化しない。
LED表示	PAUS	Prog	Prog	Prog	Prog
LCD表示	END_	例：008-END_	例：000-DO_	例：001-RE_	例：127-WAIT
P L A Y 機能	自動演奏を中断し、PAUSモードに切り替える。	受け付けない。	受け付けない。	例：127-END_	例：004-SO_
LED表示	Prog	PAUS	PLAY	PLAY	例：006-SI_
LCD表示	例：007-DO_+ END_	例：006-SI_	例：000-DO_	例：300→PLAY	例：100→PLAY
P A U S 機能	PROGモードに切り替える。データの入力は演奏アドレスの続きから行う。	自動演奏を再開し、PLAYモードに切り替える。	無音を出力する。データは記憶しない。また、アドレスも変化しない。	演奏アドレスを受け付けない。000番地に戻し、LCDは1秒間アドレスを表示する。	基本時間を+1する。LEDは基本時間を1秒間表示したあと、モード表示に戻る。
LED表示	Prog	PLAY	PAUS	PAUS	例：300→PAUS
LCD表示	例：003-FA_	例：004-SO_	例：RA#1 END_	例：000-END→(1秒後)→ END_	例：END_

備考 リセット・スタート時は、PROGモード。

11.5 キーごとの機能説明

すべてのキーは、どのキーも押されていないときに押された場合、有効になります。別のキーが押されているときは、そのキーが離されるまで、ほかのキーを押しても（多重押し）有効とはなりません。

キー・マトリクス上のキーが押されていないときは、リモコン入力を受け付けます。

キー・マトリクスの構成については、11.7 ハードウェア構成を参照してください。

次に、各キーの説明を行います。

11.5.1 MODEキー

2秒以上キーを押すと、PAUSモードとPLAYモードを交互に切り替えます。

PROGモード時は、音階キーによる演奏データの入力が可能となります。また、MODEキーを2秒以下の間押すことにより、データの入力を先頭（演奏用データ・メモリの先頭）のデータから行うか、終わり（アドレスFnH）のデータから行うかを切り替えることができます。

PLAYモードまたはPAUSモード時は、PROGモードで入力した演奏データにより、自動演奏が可能になります。また、MODEキーの二重押しにより、演奏をスタートするか、ポーズするかを切り替えることができます。

次に、MODEキーの動作と表示を示します。

モード	MODEキー		7セグメントLED (LED4, LED5, LED6, LED7)表示
	2秒以上押し	2秒以下押し	
PROGモード	PAUSモードに切り替える	データの入力を始めから行うか終わり から行うかを切り替える	
PLAYモード	PROGモードに切り替える (データ入力は終わりから)	自動演奏を中断しPAUSモードとなる	
PAUSモード (ポーズ)		自動演奏を再開しPLAYモードとなる	







リセット・スタート時はPROGモードとなります。

11.5.2 UP/DOWNキー

PROGモード時は、キーを押すごとに演奏データのアドレスをアップ／ダウンし、キーを押している間は、そのデータの音階をLCDに表示出力します。

PLAYモードまたはPAUSモード時は、キーを押すごとに基本時間Tをアップ／ダウンしてLCDに表示し、表示が最大／最小になると、約1秒後にモード表示に戻ります。

次に、UP/DOWNキーの動作および表示を示します。

モード	UPキー	DOWNキー
PROGモード 演奏データの編集 に使用	データ内容 (DO RE MI) 14セグメントLCD 	データ内容 (DO RE MI) 14セグメントLCD 
PLAYモード 基本時間の設定に 使用	7セグメントLED (Max.) T = 200 ms T = 300 ms T = 900 ms 	7セグメントLED (Min.) T = 800 ms T = 700 ms T = 100 ms 
PAUSモード 基本時間の設定に 使用	7セグメントLED (Max.) T = 200 ms T = 300 ms T = 900 ms 	7セグメントLED (Max.) T = 800 ms T = 700 ms T = 100 ms 

リセット・スタート時はT = 200 msとなります。

11.5.3 音階キー

キー・マトリクス上の、次の15のキーを音階キーといいます。

DO, RE, MI, FA, SO, RA+, SI+, DO+, DO#, RE#, FA#, SO#, RA#+, DO#+, WAIT (無音)


PROGモード、またはPAUSモード時のみ有効です。

PROGモードでは、キーを押している間、各キーに対応した音階または無音を出力し、また出力されている音階を表示します。キーが離されると、演奏用メモリ・エリアに出力した音階と音長のデータを書き込み、次のアドレスで出力される音階データを表示します。


PAUSモードでは、キーを押している間、各キーに対応した音階または無音を出力し、出力されている音階を表示します。

次に、動作例を示します。

PROGモードの場合 (14セグメントLCD)




PAUSモードの場合 (14セグメントLCD)




音階キーを押すことにより8ビットのデータを生成し、順に演奏用メモリ・エリアに読み出します。データの下位4ビットが音階データ、上位4ビットが音長データで、音長データは基本時間(T)の倍数になります。

音長データがFXHのときは、演奏を初めから繰り返すリピート動作となります。

演奏データ・メモリには最大128個までのデータが記憶できます。



逆に、演奏用メモリ・エリアからデータを取り出し、再度演奏を行うこともできます。




11.5.4 ENDキー

PROGモード時は、キーを押すごとに、リピート・コード(FnH)を書き込みます。nはキーを押している時間で決定します。

PLAYモードまたはPAUSモード時は、演奏データのアドレスをスタートに戻します。

次に、PROGモード時の動作例を示します。

14セグメントLCD



11.5.5 その他のキー

キーが押されても、何の動作も行いません。


11.6 表示説明

11.6.1 LED表示


モード表示、および基準時間Tを設定するときの時間表示に使用します。

次に、表示パターンの例を示します。


PROGモード表示



PLAYモード表示




PAUSモード表示




基本時間 (T) 表示

(300 ms設定)



数字表示パターン




11.6.2 LCD表示

演奏データ、およびアドレスの表示に使用します。

LCD表示は、LCDコントローラ／ドライバにSBIモードでキャラクタ・コードを転送することにより、行います。

次に、表示例を示します。

SI+キー・オン




“_”は空白（コード2EH）を示す。


音階表示。音階キーが押されていないときはアドレスの示すデータの内容を表示する。

アドレス表示。現在の演奏データのアドレスを示す。


WAITキー・オン



演奏データがリピート・コードのとき



11.7 ハードウェア構成



11.8 アプリケーション・プログラム

```

;***** *****
;***          ***
;***      アプリケーション・プログラム  ***
;***          ***
;***** *****

NAME    MAIN           ;モジュール名
STKLN   40            ;スタック・サイズ指定

VENT0   MBE=0, RBE=0, INIT     ;RESET
VENT1   MBE=0, RBE=1, INTBT    ;INTBT/INT4
VENT2   MBE=0, RBE=1, INTO     ;INT0
VENT4   MBE=0, RBE=1, SBI      ;SBI

DSEG0   DSEG    0      AT 20H    ;メモリ・バンク0, アドレス20Hから格納

KEYCODE: DS 2           ;キー・ソース, キー・リターンをコード化
KEYR:    DS 2           ;キー・リターン
KEYRB:   DS 2           ;
MUS:     DS 2           ;演奏データ・アドレス
KEYTIM:  DS 2           ;キー・オン時間カウント用タイマ
LCD:    DS 16          ;LCD表示用エリア
WORK:   DS 2            ;リモコン受信したデータを格納
RMDATA: DS 2            ;有効なリモコン・データ・コードを格納
KEYS:   DS 1            ;キー・ソース
KEYSB:   DS 1            ;
KEYCATT: DS 1           ;キーのチャタリング・カウンタ
KEYFLG:  DS 1           ;フラグ
TONE:   DS 1            ;音階データ
TONEFLG: DS 1           ;フラグ
B_TIM:   DS 1           ;ベース・タイマ(1.95msカウント・アップ)
B_TIMC:  DS 1           ;50ms, 100msカウント・タイマ(9.75msカウント・アップ)
DISPTIM: DS 1           ;LED表示用タイマ
TONETIM: DS 1           ;音長データ
T_TIM:   DS 1           ;基本時間
T_TIMC:  DS 1           ;基本時間用カウンタ
RPTIM:   DS 1           ;リモコン・リピート用タイマ
RPCODE:  DS 1           ;リモコン・リピート・コード・カウンタ
LEDDIG:  DS 1           ;LED表示桁番号
MODE:    DS 1           ;モード設定用 0:PROG, 1:PLAY, 2:PAUS
DEC:     DS 3           ;HEX → DEC
LCDTIM:  DS 1           ;1sカウンタ
LCDFLG:  DS 1           ;フラグ
MODESBI: DS 1           ;どのデータを転送中かを示す
LDCODE:  DS 1           ;リタード・コード・スキャン用カウンタ

```

```

MODEP:      DS     1      ;どのエッジを検出中かを示す
RMFLG:      DS     1      ;フラグ
KEYONF:    EQU   KEYFLG.3 ;1ならばキー・オン
KEYON2SF:  EQU   KEYFLG.2 ;1ならば2秒以上キーを押し続けている
HTKEYONF:  EQU   KEYFLG.1 ;1ならば本体キー・オン
KEYCHKF:   EQU   KEYFLG.0 ;1ならばキー・スキャン終了

TONE_F:    EQU   TONEFLG.0 ;1ならば音階出力を許可
TM0_F:     EQU   TONEFLG.1 ;1ならばタイマ・カウント・スタート
REFRES_F:  EQU   LCDFLG.0 ;1ならばLCDをリフレッシュ
REP_F:     EQU   RMFLG.0 ;1ならばリモコン・リピート用タイマ・カウントをスタート

;***** LED文字データ *****
LED_P:    EQU   00110001B ;"P"
LED_R:    EQU   11110101B ;"R"
LED_O:    EQU   11000101B ;"O"
LED_G:    EQU   00001001B ;"G"
LED_L:    EQU   11100011B ;"L"
LED_A:    EQU   00010001B ;"A"
LED_Y:    EQU   10001001B ;"Y"
LED_U:    EQU   10000011B ;"U"
LED_S:    EQU   01001001B ;"S"

LED_0:    EQU   11000101B ;"0"
LED__:   EQU   11111111B ;"__"

;***** KEYRデータ *****
KEYR_1:   EQU   11111110B
KEYR_2:   EQU   11111101B
KEYR_3:   EQU   11111011B
KEYR_4:   EQU   11110111B
KEYR_5:   EQU   11101111B
KEYR_6:   EQU   11011111B
KEYR_7:   EQU   10111111B
KEYR_8:   EQU   01111111B

;***** MAIN ROUTINE *****
MAIN_C CSEG INBLOCK

INIT:
;*****
; 初期設定
;*****
DI          ;すべての割り込みを禁止
CLR1      MBE ;MBE←0

```

```

;+++++
; システム・クロック設定
;+++++
MOV A, #0011B
MOV PCC, A ;0.95 μs, 通常動作モード

;+++++
; スタック・ポート設定
;+++++
MOV XA, #00H
MOV SP, XA ;スタック・ポート←00H
MOV SBS, A ;メモリ・パンク←0, MkIIモード

;+++++
; 入出力ポート設定
;+++++
MOV A, #0H
OUT PORT2, A
OUT PORT8, A

MOV A, #0FH
OUT PORT3, A
OUT PORT4, A
OUT PORT5, A
OUT PORT6, A
OUT PORT7, A

MOV XA, #0FH
MOV PMGA, XA ;ポート3を出力モード, ポート6を入力モードに

MOV XA, #34H
MOV PMGB, XA ;ポート2, 4, 5を出力モード, ポート7を入力モードに

MOV XA, #00H
MOV PMGC, XA ;ポート8を入力モードに

MOV XA, #0C1H ;ポート1, 2, 3は内蔵フルアップ抵抗の接続を指定しない
MOV POGA, XA ;ポート0, 6, 7は内蔵フルアップ抵抗の接続を指定する

MOV XA, #0H
MOV POGB, XA ;ポート8は内蔵フルアップ抵抗の接続を指定しない

;+++++
; ハート・ウェア設定
;+++++
MOV XA, #00H
MOV WM, XA ;時計モードを禁止
MOV TMO, XA ;タイマ/イベント・カウンタの発振を停止
MOV TM1, XA ;タイマ・カウンタの発振を停止

```

```

SET1    TOE0          ;出力許可フラグ←1
MOV     A, #0000B
MOV     IPS, A         ;高位の割り込みを設定
MOV     XA, #8AH
MOV     CSIM, XA       ;シリアル動作モードを設定
SET1    RELT

;+++++RAMクリア設定+++++
;+
;+
SET1    MBE           ;MBE←1
SEL     MBO           ;メモリ・バンク←0
MOV     A, #0H
MOV     HL, #04H

L0OP1:
MOV     @HL, A
INCS   L
BR     L0OP1

INCS   H
BR     L0OP1

SEL     MB1           ;メモリ・バンク←1
MOV     A, #0FH

L0OP2:
MOV     @HL, A
INCS   L
BR     L0OP2

INCS   H
BR     L0OP2

SEL     MBO           ;メモリ・バンク←0
CLR1   MBE           ;MBE←0

;+++++RAM設定+++++
;+
;+
MOV     A, #0DH
MOV     LDPCODE, A      ;LDPCODEを初期化
MOV     A, #0111B
MOV     LEDDIG, A        ;LEDDIGを初期化
MOV     A, #2H
MOV     T_TIM, A         ;T_TIMを初期化

;+++++ベーシック・インターバル・タイマ設定+++++
;+
;+
MOV     A, #1111B
MOV     BTM, A           ;タイマ・スタート

```

```

EI      IEBT           ;ペ-シック・インターバル・タイマ割り込み(INTBT)を許可
EI      ;すべての割り込みを許可

;*****
;      メイン・ルーチン
;*****

MAIN:
    CALL    !RMDEC          ;リモコン・コード処理
    CALL    !KEYDEC          ;キー・コード
    CALL    !LCDDATA         ;LCD表示データ設定
    CALL    !TONEOUT         ;メロディ出力
    CALL    !TIM              ;タイマ処理
    BR     MAIN

;***** TABLE LOOK UP *****
$IC=TABLE.INC

;***** SUBROUTINESE *****
$IC=SUB.INC

;+++++
;      リモコン・コード処理
;+++++

RMDEC:
    SKF    HTKEYONF        ;1ならば本体キー・オン
    RET

    SKTCLR IRQT1           ;1ならばモジュロ・レジ・スタ=カウント・レジ・スタ
    BR     MODEP7
    MOV    A, #0H
    MOV    MODEP, A          ;MODEP←0H
    MOV    A, #0DH
    MOV    LDCODE, A          ;LDCODEを初期化
    DI     IEO               ;INT0割り込みを禁止
    RET

MODEP7:
    MOV    A, MODEP
    SKE    A, #7H             ;MODEP=7?
    RET
    MOV    A, #0H
    MOV    MODEP, A          ;MODEP←0H
    MOV    RPCODE, A          ;リピート・コード回数をクリア
    MOV    RPTIM, A          ;200msカウントをクリア
    MOV    XA, RMDATA
    MOV    HL, XA
    ADDS   XA, #0EDH
    BR     MODEP7_1
    RET

```

```

MODEP7_1:
    MOV     XA, HL
    MOV     BC, #HIGH  RM_T
    CALL   !TABLE
    MOV     BC, XA          ;BCレジスタ←RMDATAをコード化
    SKF     KEYONF
    RET
    SET1   KEYONF
    MOV     XA, #10H
    SKE     XA, BC
    BR     MODEP7_2
    CLR1   KEYON2SF
    MOV     XA, #0H
    MOV     KEYTIM, XA

MODEP7_2:
    MOV     XA, BC
    MOV     KEYCODE, XA
    SET1   KEYCHKF
    RET

;+++++キー・コード+
;キー・コード
;+++++キー・コード+

KEYDEC:
    SKT     KEYCHKF      ;1ならばキー・スキャンを終了
    RET
    CLR1   KEYCHKF
    SKT     KEYONF       ;1ならばキー・オン
    BR     KEYOFF
    MOV     XA, KEYCODE
    ADDS   XA, #0FOH      ;音階キー・オン?
    BR     ONKA11
    ADDS   A, #0FH
    BR     MODE1         ;0ならばMODEキー・オン
    ADDS   A, #0FH
    BR     UPKEY1        ;1ならばUPキー・オン
    ADDS   A, #0FH
    BR     DOWN1         ;2ならばDOWNキー・オン
    BR     ENDKEY1       ;3ならばENDキー・オン

;-----音階キー・オン-----
;-----音階キー・オン-----

ONKA11:
    MOV     A, MODE      ;PROGモード?
    SKE     A, #0H
    BR     ONKA11_1
    MOV     A, #0H          ;音長データ←0H

```

```

MOV    T_TIMC,A
BR    ONKAI1_2

ONKAI1_1:
SKE    A,#2H           ;PAUSEモード?
RET
MOV    A,#0FH          ;音長データ←FH
ONKAI1_2:
MOV    TONETIM,A
MOV    A,KEYCODE        ;TONE←KEYCODE
MOV    TONE,A
SET1   TONE_F
RET

;-----;
; MODEキー(≥2s)
;-----;

MODE1:
SKT    KEYON2SF
RET
MOV    A,MODE
SKE    A,#0H           ;PROGモード?
BR    MODE1_1
MOV    A,#2H
MOV    MODE,A          ;MODE←PAUSEモード
RET

MODE1_1:
MOV    A,#0H
MOV    MODE,A          ;MODE←PROGモード
CLR1   TONE_F
RET

;-----;
; UPキー・オン
;-----;

UPKEY1:
MOV    A,MODE
SKE    A,#0H           ;PROGモード?
BR    UPKEY1_1
MOV    XA,MUS
ADDS   XA,#2H
NOP
MOV    MUS,XA          ;演奏データ・アドレス+2H
BR    DOWN1_1

UPKEY1_1:
MOV    A,T_TIM
SKE    A,#9H
INCS   T_TIM          ;基本時間+1H

```

```

NOP
BR      DOWN1_3

;-----;
;    DOWNキー・オン
;-----;

DOWN1:
MOV    A, MODE
SKE    A, #0H          ;PROGモード?
BR    DOWN1_2
MOV    XA, MUS
DECS   XA
NOP
DECS   XA
NOP
MOV    MUS, XA          ;演奏データ・アドレス-2
DOWN1_1:
MOV    XA, MUS
MOV    HL, XA
INCS   HL
NOP
SET1   MBE             ;MBE←1
SEL    MB1             ;メモリ・バンク←1
MOV    A, @HL
NOP
SEL    MBO             ;メモリ・バンク←0
CLR1   MBE             ;MBE←0
MOV    TONE, A          ;音階データ←@演奏データ・アドレス+1
SET1   TONE_F           ;キー・オンの間演奏データを出力
RET

DOWN1_2:
MOV    A, T_TIM
SKE    A, #1H
DECS   A
NOP
MOV    T_TIM, A          ;基本時間-1

DOWN1_3:
MOV    A, #0AH
MOV    DISPTIM, A          ;基本時間表示のかウント・スタート
RET

```

```

;-----  

; ENDキー・オン  

;-----  

ENDKEY1:  

    MOV     A, MODE  

    SKE     A, #0H          ;PROGモード?  

    BR      END2  

    MOV     A, #0H  

    MOV     TONETIM, A      ;音長カウント・スタート  

    MOV     T_TIMC, A  

    BR      ONKA12_1  

END2:  

    CLR1   TONE_F  

    MOV     A, MODE  

    SKE     A, #1H          ;PLAYモード?  

    BR      END2_2  

    MOV     A, #0H  

    MOV     TONETIM, A  

    MOV     T_TIMC, A  

END2_1:  

    MOV     XA, #0H  

    MOV     MUS, XA          ;演奏データ・アドレス←0H  

    RET  

END2_2:  

    MOV     A, #0AH  

    MOV     LCDTIM, A        ;LCD表示用タイマ(1s)のかウント・スタート  

    BR      END2_1  

;-----  

; キー・オフ  

;-----  

KEYOFF:  

    MOV     XA, KEYCODE  

    ADDS   XA, #0F0H          ;音階キー・オフ?  

    BR      ONKA12  

;-----  

; ENDキー・オフ  

;-----  

    SKE     A, #3H          ;ENDキー・オフ?  

    BR      MODE2  

    MOV     A, MODE  

    SKE     A, #0H          ;PROGモード?  

    RET  

    MOV     XA, MUS  

    MOV     HL, XA  

    MOV     A, TONETIM  

    MOV     B, A

```

```

SET1    MBE          ;MBE←1
SEL     MB1          ;メモリ・バンク←1
MOV     A, #0FH
MOV     @HL, A        ;@演奏データ・アドレス←FH
INC8   HL           ;演奏データ・アドレス+1H
NOP
MOV     A, B
MOV     @HL, A        ;@演奏データ・アドレス+1←音長データ
SEL     MBO          ;メモリ・バンク←0
CLR1   MBE          ;MBE←0
RET

;-----  

; 音階キー・オフ  

;-----  

ONKA12:
MOV     A, MODE
SKE    A, #0H          ;PROGモード?
BR    ONKA12_2
MOV     A, TONETIM
CALL   !TONE_S          ;Aレジスタ←TONETIM
MOV     XA, MUS
ADDS   XA, #2H
NOP
MOV     MUS, XA        ;演奏データ・アドレス+2H
INC8   HL           ;演奏データ・アドレス+1H
NOP
MOV     A, TONE          ;Aレジスタ←TONE

SET1    MBE          ;MBE←1
SEL     MB1          ;メモリ・バンク←1
MOV     @HL, A        ;@演奏データ・アドレス+1H←TONE
SEL     MBO          ;メモリ・バンク←0
CLR1   MBE          ;MBE←0

ONKA12_1:
CLR1   TONE_F         ;無音を出力
RET

ONKA12_2:
SKE    A, #2H          ;PAUSEモード?
RET
BR    ONKA12_1

```

```

;-----  

; MODEキー(<2s)  

;-----  

MODE2:  

    SKE    A, #0H          ; MODEキー・オフ?  

    BR     ONKA12_1  

    SKF    KEYON2SF  

    RET  

    MOV    A, MODE  

    SKE    A, #0H          ; PROGモード?  

    BR     MODE2_4  

    MOV    XA, MUS  

    ADDS   XA, #0FFH       ; 演奏データ・アドレス=00H?  

    BR     MODE2_2  

    BR     END2_1  

MODE2_1:  

    MOV    MUS, XA  

MODE2_2:  

    MOV    XA, MUS  

    MOV    HL, XA  

    SET1   MBE  

    SEL    MB1  

    MOV    A, @HL  

    SEL    MB0  

    CLR1   MBE  

    SKE    A, #0FH          ; @演奏データ・アドレス=FH?  

    BR     MODE2_3  

    RET  

MODE2_3:  

    MOV    XA, MUS  

    ADDS   XA, #2H          ; 演奏データ・アドレス+2H>256?  

    BR     MODE2_1  

    BR     END2_1  

MODE2_4:  

    SKE    A, #1H          ; PLAYモード?  

    BR     MODE2_5  

    MOV    A, #2H  

    MOV    MODE, A          ; MODE←PAUSEモード  

    BR     ONKA12_1  

MODE2_5:  

    MOV    A, #1H  

    MOV    MODE, A          ; MODE←PLAYモード  

    MOV    A, #0H

```

```

MOV      TONETIM,A
MOV      T_TIMC,A

;+++++LCD表示データ設定+++++
;LCDDATA:
SKT      REFRES_F          ;1ならばLCDをリフレッシュ
RET
CLR1     REFRES_F          ;REFRES_F←0
;-----アドレスHEX-->DEC変換-----
MOV      A, #0H
MOV      DEC, A              ;DEC←0H
MOV      A, MUS
XCH      A, E                ;Eレジスタ←MUS

MOV      A, MUS+1
LCDHEX1:
MOV      HL, #DEC+1          ;10進補正(1桁目)
ADDS   A, #6H
BR      LCDHEX4

MOV      @HL, A              ;2桁目をインクリメント
INCS   L
MOV      A, #7H
ADDS   A, @HL              ;10進補正
BR      LCDHEX4

INCS   DEC                  ;3桁目をインクリメント
LCDHEX2:
MOV      @HL, A              ;2桁目の値の回数に,6を加算
DECS   E
BR      LCDHEX3
BR      LCD1

LCDHEX3:
MOV      A, DEC+2
ADDS   A, #6H
BR      LCDHEX1

LCDHEX4:
ADDS   A, #0AH
NOP
BR      LCDHEX2

LCD1:
MOV      A, MODE
SKE      A, #2H              ;PAUSEモード?

```

```

BR      LCD2
MOV    A, LCDTIM
SKE    A, #0H           ;LCD表示用タイマ(1s)=0?
BR      LCD2
MOV    XA, #2EH
MOV    LCD, XA          ;アドレスの1行目を設定
MOV    LCD+2, XA        ;アドレスの2行目を設定
MOV    LCD+4, XA        ;アドレスの3行目を設定
BR      LCD3

LCD2:
MOV    X, #0H
MOV    A, DEC
MOV    LCD, XA          ;アドレスの1行目を設定
MOV    A, DEC+1
MOV    LCD+2, XA        ;アドレスの2行目を設定
MOV    A, DEC+2
MOV    LCD+4, XA        ;アドレスの3行目を設定

LCD3:
MOV    XA, #27H
MOV    LCD+6, XA        ;”-”データ設定
MOV    XA, MUS
MOV    HL, XA
MOV    XA, @HL          ;XAレジスタ←@演奏データ・アドレス
ADDS  XA, XA
ADDS  XA, XA          ;XAレジスタ×4
MOV    DE, XA
MOV    BC, #HIGH LCD_T
CALL  !TABLE
MOV    LCD+8, XA        ;音階データの1行目を設定
CALL  !TABLEINC
MOV    LCD+10, XA       ;音階データの2行目を設定
CALL  !TABLEINC
MOV    LCD+12, XA       ;音階データの3行目を設定
CALL  !TABLEINC
MOV    LCD+14, XA       ;音階データの4行目を設定
SKT   PORT0.2          ;SB0=ハイ・レベル?
RET
SKT   PORT0.1          ;SCK=ハイ・レベル?
RET
SET1  CMDT             ;CMDTセット
NOP
NOP
SET1  RELT              ;RELTセット
NOP
NOP
SKT   PORT0.2          ;SB0=ハイ・レベル?
RET
SET1  CMDT             ;CMDTセット
MOV    XA, #01H

```

```

MOV    S10, XA           ;シフト・レジスタ←スレーブ・アドレス(01H)
MOV    A, #1H
MOV    MODESBI, A        ;MODESBI←1H
EI    IECSI              ;SBI割り込みを許可
MOV    XA, #1H
MOV    SVA, XA            ;スレーブ・アドレス・レジスタ←1H
CLR1  REFRES_F          ;REFRES_F←0
RET

;+++++メロディ出力+++++
;TONEOUT:
SKT    TONE_F            ;1ならば音階出力を許可
BR     TONE0
MOV    X, #0H
MOV    A, TONE
MOV    BC, #HIGH TONE_T
CALL   !TABLE
MOV    TMODO, XA          ;モジュロ・レジスタ←音階データ
MOV    A, TONE
ADDS  A, #2H
BR     TONE1

TONE0:
MOV    XA, #0H             ;タイマ・カウントを停止
CLR1  TM0_F              ;0ならばタイマ・カウント・スタート
BR     TONE2

TONE1:
SKF    TM0_F
RET
MOV    XA, #7CH             ;タイマ・カウント・スタート
SET1  TM0_F              ;1ならばタイマ・カウントを停止

TONE2:
MOV    TM0, XA
RET

;+++++タイマ処理+++++
;TIM:
MOV    A, B_TIMC
SKE    A, #0AH              ;B_TIMC≥100ms?
BR     TIM_50

;=====100ms通過=====
MOV    A, #0H
MOV    B_TIMC, A
SKT    REP_F

```

```

BR      TIM_51
INCS   RPTIM           ;リモコン・リピート用タイマ+1
NOP
MOV    A, RPTIM
SKE    A, #2H           ;リモコン・リピート用タイマ=200ms?
BR      TIM_51
MOV    A, #0H
MOV    RPTIM, A         ;RPTIMをクリア
MOV    A, RPCODE
ADDS   A, #0FH          ;リモコン・リピート・コード・カウンタ=0
BR      TIM_52
ADDS   A, #0FH          ;リモコン・リピート・コード・カウンタ=1
BR      TIM_54
ADDS   A, #0FH          ;リモコン・リピート・コード・カウンタ=2
BR      TIM_54
ADDS   A, #0FH          ;リモコン・リピート・コード・カウンタ=3
BR      TIM_54

TIM_52:
SKF    HTKEYONF
BR      TIM_51
SET1   KEYCHKF         ;リモコン・リピート・コード・カウンタ=0 or ≥4
CLR1   KEYONF           ;KEYONF←0
CLR1   REP_F            ;REP_F←0

TIM_54:
MOV    A, #0H
MOV    RPCODE, A         ;RPCODEをクリア

TIM_51:
MOV    XA, KEYCODE
MOV    HL, #10H
SKE    XA, HL
BR      TIM_101
SKF    KEYON2SF
BR      TIM_101
MOV    XA, KEYT1M
MOV    HL, #14H
SKE    XA, HL           ;キーを2秒以上押し続けている?
BR      TIM_105
SET1   KEYON2SF         ;1ならばキーを2秒以上押し続けている
SET1   KEYCHKF

TIM_101:
MOV    A, DISPTIM
SKE    A, #0H             ;LED表示用タイマ(1s)をカント中?
BR      TIM_107

TIM_102:
MOV    A, LCDTIM
SKE    A, #0H             ;LCD表示用タイマ(1s)をカント中?
BR      TIM_108

TIM_103:
MOV    A, MODE

```

```

SKE    A, #0H          ;PROGモード?
BR     TIM_109
INCS   T_TIMC          ;基本時間用カウント+1
NOP
MOV    A, T_TIM
MOV    L, A
MOV    A, T_TIMC
SKE    A, L            ;基本時間=基本時間用カウント?
BR     TIM_53
MOV    A, #0H
MOV    T_TIMC, A
MOV    A, TONETIM
SKE    A, #0EH          ;音長データ=EH?
BR     TIM_104
BR     TIM_53

TIM_104:
INCS   TONETIM         ;音長データ+1
NOP
BR     TIM_53

;-----  

;   キー・オン時間カウント  

;-----  

;-----  

TIM_105:
MOV    XA, KEYTIM
MOV    HL, XA
INCS   HL
NOP
MOV    XA, HL
MOV    KEYTIM, XA        ;キー・オン時間カウント用タイマ+1
BR     TIM_101

;-----  

;   LED表示用タイマ(1s)カウント  

;-----  

;-----  

TIM_107:
MOV    A, DISPTIM
DECS   A
NOP
MOV    DISPTIM, A        ;LED表示用(1s)タイマ-1
BR     TIM_102

;-----  

;   LCD表示用タイマ(1s)カウント  

;-----  

;-----  

TIM_108:
MOV    A, LCDTIM
DECS   A
NOP

```

```

MOV    LCDTIM,A      ;LCD表示用(1s)タイマ-1
BR     TIM_103

TIM_109:
SKE    A,#1H          ;PLAYモード?
BR     TIM_53
INCS   T_TIMC         ;基本時間用カウント+1
NOP

TIM_110:
MOV    A,T_TIM
MOV    L,A
MOV    A,T_TIMC
SKE    A,L            ;基本時間=基本時間用カウント?
BR     TIM_53
MOV    A,#0H
MOV    T_TIMC,A
MOV    A,TONETIM
DECS   A
BR     TIM_116
MOV    XA,MUS

SET1   MBE             ;MBE←1
SEL    MB1             ;メモリ・バンク←1
MOV    HL,XA
MOV    A,@HL+
NOP
MOV    B,A
MOV    A,@HL
MOV    C,A
SEL    MBO             ;メモリ・バンク←0
CLR1   MBE             ;MBE←0

MOV    A,B
MOV    TONETIM,A       ;音長データ←@演奏データ・アドレス
MOV    A,C
MOV    TONE,A           ;音階データ←@演奏データ・アドレス+1
MOV    A,#0H
MOV    T_TIMC,A
SKE    C,#0FH
BR     TIM_115
BR     TIM_114

TIM_115:
SKE    B,#0FH
BR     TIM_111
BR     TIM_112

TIM_112:
MOV    XA,#0H
MOV    MUS,XA

```

```

MOV    A, TONE
MOV    TONETIM, A      ;TONETIM←リピート時間
MOV    A, #0FH
MOV    TONE, A          ;TONE←FH
TIM_114:
CLR1   TONE_F
SKE    B, #0FH
BR     TIM_113
BR     TIM_53

TIM_111:
SET1   TONE_F

TIM_113:
MOV    XA, MUS
ADDS  XA, #2H
NOP
MOV    MUS, XA          ;演奏データ・アドレス+2
BR     TIM_53

TIM_116:
MOV    TONETIM, A      ;音長データ-1
BR     TIM_53

;=====
; 50ms通過
;=====

TIM_50:
SKE    A, #5H          ;B_TIME≥50ms?
RET

;-----リモコン・リピート用タイマ・カウント-----
;-----

TIM_53:
SET1   REFRES_F        ;1ならばLCD表示をリフレッシュ
RET

;+++++
;リモコン処理
;+++++
INTO:
DI
PUSH   BS
SEL    RB1              ;レジスタ・バッファ←1
MOV    A, MODEP
ADDS  A, #0FH            ;MODEP=0?
BR    INTO_E
ADDS  A, #0FH            ;MODEP=1?
BR    INTO_P1
ADDS  A, #0FH            ;MODEP=2?
BR    INTO_P2

```

```

ADDS    A, #09H          ;MODEP=3-6?
BR     INTO_PX
BR     INTO_E

INTO_P1:
MOV    A, #2H
MOV    MODEP, A          ;MODEP←2
MOV    XA, #62H
MOV    TMOD1, XA          ;モジュロ・レジスタ←6ms
MOV    A, #1H
MOV    IMO, A          ;立ち下がりイッジを指定
BR     INTO_E

INTO_P2:
MOV    XA, T1
ADDS  XA, #0C9H          ;カウント・レジスタ=3.375ms?
BR     INTO_RP1
MOV    A, #3H
MOV    MODEP, A          ;MODEP←3
BR     INTO_W1

INTO_RP1:
INCS  RPCODE           ;リモコン・リピート用コード・カウンタ+1
NOP
BR     INTO_P0

INTO_PX:
MOV    XA, T1
ADDS  XA, #0E4H          ;カウント・レジスタ=1.6875ms?
BR     INTO_CY0
SET1  CY                ;CY←1
BR     INTO_L1

INTO_CY0:
CLR1  CY                ;CY←0
INTO_L1:
MOV    XA, WORK
MOV    B, A
MOV    A, X
RORC  A
MOV    X, A
MOV    A, B
RORC  A
MOV    WORK, XA           ;受信データ(WORK)を右へ1ビット・シフト
SKT   CY
BR     INTO_E
MOV    A, MODEP
ADDS  A, #0CH            ;MODEP=3?
BR     INTO_P3
ADDS  A, #0FH            ;MODEP=4?

```

```

BR    INTO_P4
ADD$ A, #0FH           ;MODEP=5?
BR    INTO_P5
MOV   XA, WORK
MOV   HL, #0FFH
XOR   HL, XA
MOV   XA, RMDATA
SKE   XA, HL           ;有効なリモコン・データ=受信データ反転?
BR    INTO_P0
MOV   A, #7H
MOV   MODEP, A          ;MODEP←7
SET1  REP_F
DI    IEO               ;INT0割り込みを禁止
MOV   XA, #0H
MOV   TM1, XA           ;タイマ・カウンタを停止
BR    INTO_E1

INT0_P3:
MOV   XA, WORK
MOV   HL, #00H
SKE   XA, HL           ;カスタム・コード=00H?
BR    INTO_P0
MOV   A, #4H
MOV   MODEP, A          ;MODEP←4
BR    INTO_W1

INT0_P4:
MOV   XA, WORK
MOV   HL, #0FFH
SKE   XA, HL           ;カスタム・コード=FFH?
BR    INTO_P0
MOV   A, #5H
MOV   MODEP, A          ;MODEP←5
BR    INTO_W1

INT0_P5:
MOV   XA, WORK
MOV   RMDATA, XA        ;有効なリモコン・データ←受信データ
MOV   A, #6H
MOV   MODEP, A          ;MODEP←6
BR    INTO_W1

INT0_P0:
MOV   A, #0H
MOV   MODEP, A          ;MODEP←0
MOV   A, #0DH
MOV   LDCODE, A          ;LDCODEを初期化
DI    IEO               ;INT0割り込みを禁止
MOV   XA, #0H
MOV   TM1, XA           ;タイマ・カウンタを停止
BR    INTO_E1

```

```

INT0_W1:
    MOV     XA, #42H
    MOV     TMOD1, XA      ;モジュロ・レジスタ←4ms
    MOV     XA, #80H
    MOV     WORK, XA      ;WORKを初期化

INT0_E:
    MOV     XA, #6CH
    MOV     TM1, XA      ;タイマ・カウント・スタート

INT0_E1:
    POP    BS
    EI
    RETI      ;すべての割り込みを許可

;+++++
;      タイマ
;+++++
INTBT:
    SEL    RB2      ;レジスタ・バンク←2
    DI     IECSI    ;SBI割り込みを禁止
    EI
    CALL   !TIMCNT  ;すべての割り込みを許可
    CALL   !RMCNT   ;タイマ・カウント
    CALL   !KEYCHK  ;リモコン処理カウント
    CALL   !LED1     ;キー・チェック
    CALL   !IECSI    ;LED表示
    EI
    RETI      ;SBI割り込みを許可

;+++++
;      SBI処理
;+++++
SBI:
    SEL    RB3      ;レジスタ・バンク←3
    DI     IEBT     ;ペースィック・インターバル・タイマ割り込み(INTBT)を禁止
    EI
    MOV    A, MODESBI ;すべての割り込みを許可
    ADDS  A, #0FH
    BR    SB1_E
    ADDS  A, #0FH
    BR    SB11
    ADDS  A, #0FH
    BR    SB12
    ADDS  A, #0FH
    BR    SB13
    ADDS  A, #0FH
    BR    SB14
    ADDS  A, #0FH
    BR    SB15
    ADDS  A, #0FH
    BR    SB16

```

```

ADD$ A, #0FH
BR SB17
ADD$ A, #0FH
BR SB18
ADD$ A, #0FH
BR SB19
ADD$ A, #0FH
BR SB110
ADD$ A, #0FH
BR SB111
BR SB1_E

SB11:
MOV BC, #04H ;シフト・レジ「スタ←ホ」ンタ・コマンド (04H)
BR SB1TIM

SB12:
MOV BC, #07H ;シフト・レジ「スタ←ホ」ンタ(07H)
BR SB1TIM

SB13:
MOV BC, #01H ;シフト・レジ「スタ←ライト・コマンド」 (01H)
BR SB1TIM

SB14:
MOV XA, LCD ;シフト・レジ「スタ←アト」レス1桁目
MOV BC, XA
BR SB1TIM

SB15:
MOV XA, LCD+2 ;シフト・レジ「スタ←アト」レス2桁目
MOV BC, XA
BR SB1TIM

SB16:
MOV XA, LCD+4 ;シフト・レジ「スタ←アト」レス3桁目
MOV BC, XA
BR SB1TIM

SB17:
MOV XA, LCD+6 ;シフト・レジ「スタ←」_「テ」-タ
MOV BC, XA
BR SB1TIM

SB18:
MOV XA, LCD+8 ;シフト・レジ「スタ←音階テ」-タ1桁目
MOV BC, XA
BR SB1TIM

```

```

SB19:
    MOV     XA,LCD+10      ;シフト・レジスタ←音階データ2桁目
    MOV     BC,XA
    BR     SBITIM

SBI10:
    MOV     XA,LCD+12      ;シフト・レジスタ←音階データ3桁目
    MOV     BC,XA
    BR     SBITIM

SBI11:
    MOV     XA,LCD+14      ;シフト・レジスタ←音階データ4桁目
    MOV     BC,XA

SBITIM:
    MOV     L,#1110B        ;100 μsカウントを初期化

SBICNT:
    DECS   L
    BR     SBI_ACKD

SB10:
    MOV     A,#0H
    MOV     MODESBI,A      ;MODESBI←0
    SET1   REFRES_F        ;1ならばLCD表示をリフレッシュ
    BR     SBI_E

SBI_ACKD:
    SKT     ACKD          ;ACKリッジ検出フラグ=1?
    BR     SBICNT
    SKT     C0I            ;アドレス・コンバーティ信号=1?
    BR     SB10
    SKT     PORT0.2        ;SB0=ハイ・レベル?
    BR     SBI_E
    SKT     PORT0.1        ;SCK=ハイ・レベル?
    BR     SBI_E
    MOV     A,MODESBI
    ADDS   A,#0FH          ;MODESBI=0?
    BR     SBI_S10
    ADDS   A,#0FH          ;MODESBI=1?
    BR     SBI_CMDT
    ADDS   A,#0FH          ;MODESBI=2?
    BR     SBI_S10
    ADDS   A,#0FH          ;MODESBI=3?

SBI_CMDT:
    SET1   CMDT           ;CMDTセット

SBI_S10:
    MOV     XA,BC
    MOV     S10,XA          ;シフト・レジスタ←BCレジスタ
    INCs   MODESBI          ;MODESBI+1
    NOP
    SET1   CSIE            ;シリアル・インターフェース動作, シフト・レジスタを禁止
    EI     IECSI           ;SBI割り込みを許可

```

```
SBI_E:  
EI      IEBT      ;ペースック・インターバル・タイマ割り込み(INTBT)を許可  
RETI  
  
END
```

```

$NOLIST
;*****
;***          サブルーチン          ***
;*****
$LIST
$TT='SUBROUTINE'
TONE_S:
;+++++
;+++    音長データ      ++
;+++++
        MOV    B,A
        MOV    XA,MUS
        MOV    HL,XA
        SET1   MBE           ;MBE←1
        SEL    MB1           ;メモリ・パンク←1
        MOV    A,B
        MOV    @HL,A
        CLR1   MBE           ;MBE←0
        RET

;+++++
;     タイマ・カウント
;+++++
TIMCNT:
        INCs   B_TIM          ;ベース・タイマ+1
        NOP
        MOV    A,B_TIM
        ADDS   A,#0BH          ;ベース・タイマ>10ms?
        RET
        INCs   B_TIMC         ;100msカウント用タイマ+1
        NOP
        MOV    A,#0H
        MOV    B_TIM,A          ;ベース・タイマ←0H
        RET

;+++++
;     リモコン処理カウント
;+++++
RMCNT:
        SKF    HTKEYONF
        BR    RMCNT_1
        MOV    A,MODEP
        SKE    A,#0H           ;MODEP=0?
        RET
        SKF    PORT1.0
        BR    RMCNT_1
        INCs   LDCODE          ;リターン・コード・スキャン用カウンタ+1
        RET
        MOV    A,#0DH          ;LDCODEを初期化

```

```

MOV    LDCODE, A
MOV    A, #1H
MOV    MODEP, A          ;MODEP←1
MOV    A, #0H
MOV    IMO, A            ;立ち上がりエッジを指定
MOV    XA, #62H           ;モジュロ・レジスタ←6ms
MOV    TMOD1, XA
MOV    XA, #6CH
MOV    TM1, XA            ;タイマ・カウント・スタート
CLR1   IRQ0
EI     IEO               ;INT0割り込みを許可
RET

RMCNT_1:
MOV    A, #0DH
MOV    LDCODE, A          ;LDCODEを初期化
RET

;+++++
;キー・チェック
;+++++
KEYCHK:
MOV    A, LEDDIG
SKE    A, #0111B          ;LEDDIG=0111B?
RET
MOV    A, #0H
OUT   PORT3, A            ;ポート3←0
MOV    XA, #0FFH
OUT   PORT4, XA           ;ポート4,5←ALL1
IN    XA, PORT6
MOV    BC, #0FFH
SKE    XA, BC              ;ポート6,7=ALL1?
BR    KEYCHK1
SKT    HTKEYONF           ;1ならば本体キー・オン
RET
SKT    KEYONF             ;1ならばキー・オン
RET
INC$  KEYCATT             ;チャタリング・カウンタ+1
NOP
MOV    A, KEYCATT
ADDS  A, #0DH              ;キーが3回一致?
RET
MOV    A, #0H
MOV    KEYCATT, A           ;チャタリング・カウンタ←0H
CLR1   KEYONF              ;KEYONF←0
SET1   KEYCHKF             ;1ならばキー・スキャンを終了
CLR1   HTKEYONF
RET

```

```

KEYCHK1:
    MOV     A, #1110B
    MOV     KEYS, A          ;キー・ソース←1110B
KEYCHK2:
    OUT    PORT3, A          ;ポート3←1110B
    NOP
    NOP
    NOP
    NOP
    NOP
    NOP
    NOP
    NOP
    NOP
    IN     XA, PORT6
    MOV     KEYR, XA          ;キー・リターン←ポート6,7
    MOV     BC, #0FFH
    SKE    XA, BC          ;ポート6,7=ALL1?
    BR     KEYCHK3
    SET1   CY              ;CY←1
    MOV     A, KEYS
    ADDC  XA, XA
    MOV     KEYS, A          ;キー・ソースを左へ1ビット・シフト
    SKE    A, #0111B
    BR     KEYCHK2
    RET

KEYCHK3:
    MOV     A, KEYS
    MOV     HL, #KEYSB
    SKE    A, @HL          ;キー・ソース=KEYSB?
    BR     KEYCHK23
    MOV     A, KEYR
    MOV     HL, #KEYRB
    SKE    A, @HL          ;キー・リターン=KEYRB?
    BR     KEYCHK23
    MOV     A, KEYR+1
    MOV     HL, #KEYRB+1
    SKE    A, @HL          ;キー・リターン+1=KEYRB+1?
    BR     KEYCHK23
    INCs   KEYCATT         ;チャタリング・カウンタ+1
    NOP
    MOV     A, KEYCATT
    ADDS  A, #0DH          ;キーが3回一致?
    RET
    MOV     A, KEYS
    SKE    A, #1110B         ;キー・ソース=1110B?
    BR     KEYCHK11
    MOV     XA, KEYR

```

```

MOV     HL, #KEYR_1
SKE     XA, HL
BR     KEYCHK4
MOV     BC, #01H
BR     KEYCHK22

KEYCHK4:
MOV     HL, #KEYR_2
SKE     XA, HL
BR     KEYCHK5
MOV     BC, #03H           ;KEYCODE←03H
BR     KEYCHK22

KEYCHK5:
MOV     HL, #KEYR_3
SKE     XA, HL
BR     KEYCHK6
MOV     BC, #06H           ;KEYCODE←06H
BR     KEYCHK22

KEYCHK6:
MOV     HL, #KEYR_4
SKE     XA, HL
BR     KEYCHK7
MOV     BC, #08H           ;KEYCODE←08H
BR     KEYCHK22

KEYCHK7:
MOV     HL, #KEYR_5
SKE     XA, HL
BR     KEYCHK8
MOV     BC, #0AH           ;KEYCODE←0AH
BR     KEYCHK22

KEYCHK8:
MOV     HL, #KEYR_6
SKE     XA, HL
BR     KEYCHK9
MOV     BC, #0DH           ;KEYCODE←0DH
BR     KEYCHK22

KEYCHK9:
MOV     HL, #KEYR_7
SKE     XA, HL
BR     KEYCHK10
MOV     BC, #0FH           ;KEYCODE←0FH
BR     KEYCHK22

KEYCHK10:
MOV    HL, #KEYR_8

```

```

SKE    XA, HL
RET
MOV    BC, #13H           ;KEYCODE←13H
BR     KEYCHK22

KEYCHK11:
SKE    A, #1101B          ;キー・ソース=1101B?
BR     KEYCHK14
MOV    XA, KEYR
MOV    HL, #KEYR_1
SKE    XA, HL
BR     KEYCHK12
MOV    BC, #00H           ;KEYCODE←00H
BR     KEYCHK22

KEYCHK12:
MOV    HL, #KEYR_7
SKE    XA, HL
BR     KEYCHK13
MOV    BC, #11H           ;KEYCODE←11H
BR     KEYCHK22

KEYCHK13:
MOV    HL, #KEYR_8
SKE    XA, HL
RET
MOV    BC, #12H           ;KEYCODE←12H
BR     KEYCHK22

KEYCHK14:
SKE    A, #1011B          ;キー・ソース=1011B?
RET
MOV    XA, KEYR
MOV    HL, #KEYR_1
SKE    XA, HL
BR     KEYCHK15
MOV    BC, #0CH            ;KEYCODE←0CH
BR     KEYCHK22

KEYCHK15:
MOV    HL, #KEYR_2
SKE    XA, HL
BR     KEYCHK16
MOV    BC, #0BH           ;KEYCODE←0BH
BR     KEYCHK22

KEYCHK16:
MOV    HL, #KEYR_3
SKE    XA, HL
BR     KEYCHK17

```

```

MOV BC, #09H ;KEYCODE←09H
BR KEYCHK22

KEYCHK17:
MOV HL, #KEYR_4
SKE XA, HL
BR KEYCHK18
MOV BC, #07H ;KEYCODE←07H
BR KEYCHK22

KEYCHK18:
MOV HL, #KEYR_5
SKE XA, HL
BR KEYCHK19
MOV BC, #05H ;KEYCODE←05H
BR KEYCHK22

KEYCHK19:
MOV HL, #KEYR_6
SKE XA, HL
BR KEYCHK20
MOV BC, #04H ;KEYCODE←04H
BR KEYCHK22

KEYCHK20:
MOV HL, #KEYR_7
SKE XA, HL
BR KEYCHK21
MOV BC, #02H ;KEYCODE←02H
BR KEYCHK22

KEYCHK21:
MOV HL, #KEYR_8
SKE XA, HL
RET
MOV BC, #10H ;KEYCODE←10H

KEYCHK22:
SKF HTKEYONF
RET
SET1 HTKEYONF ;1ならば本体キー・オン
SET1 KEYONF
MOV XA, #10H
SKE XA, BC
BR KEYCHK24
CLR1 KEYON2SF
MOV XA, #0H
MOV KEYT1M, XA

KEYCHK24:
MOV XA, BC
MOV KEYCODE, XA

```

```

SET1    KEYCHKF          ;1ならばキー・スキャンを終了
RET

KEYCHK23:
MOV     A, KEYS
MOV     KEYSB, A          ;KEYSB←キー・ソース
MOV     XA, KEYR
MOV     KEYRB, XA          ;KEYRB←キー・リターン
MOV     A, #0H
MOV     KEYCATT, A         ;チャタリング・カウンタ←0
RET

;+++++LED表示+++++
;LED1:
SET1    CY
MOV     A, LEDDIG
RORC   A                  ;LEDDIGを右へ1ビット・シフト
MOV     LEDDIG, A
SKT    CY
BR     LED2

LEDMODE0:
MOV     A, MODE
SKE    A, #0H
BR     LEDMODE1
SKT    LEDDIG.3           ;"PROG"モードを表示
BR     LED4P
SKT    LEDDIG.2
BR     LED5R
SKT    LEDDIG.1
BR     LED60
MOV     HL, #LED_G         ;LED7←"G"を表示
LED4P:
MOV     HL, #LED_P         ;LED4←"P"を表示
LED5R:
MOV     HL, #LED_R         ;LED5←"R"を表示
LED60:
MOV     HL, #LED_0         ;LED6←"0"を表示
BR     LED_OUT

LED2:
MOV     A, #0111B
MOV     LEDDIG, A          ;LEDDIGを初期化
BR     LEDMODE0

LEDMODE1:
SKE    A, #1H
BR     LEDMODE2

```

```

SKT    LEDDIG.3      ;"PLAY"モードを表示
BR     LED4P1
SKT    LEDDIG.2
BR     LED5L
SKT    LEDDIG.1
BR     LED6A
MOV    HL, #LED_Y   ;LED7←"Y"を表示
LED4P1:
MOV    HL, #LED_P   ;LED4←"P"を表示
LED5L:
MOV    HL, #LED_L   ;LED5←"L"を表示
LED6A:
MOV    HL, #LED_A   ;LED6←"A"を表示
BR     LED_CNT

LEDMODE2:
SKT    LEDDIG.3      ;"PAUS"モードを表示
BR     LED4P1
SKT    LEDDIG.2
BR     LED5A
SKT    LEDDIG.1
BR     LED6U
MOV    HL, #LED_S   ;LED7←"S"を表示
LED5A:
MOV    HL, #LED_A   ;LED5←"A"を表示
LED6U:
MOV    HL, #LED_U   ;LED6←"U"を表示
LED_CNT:
MOV    A, DISPTIM
ADDS  A, #0FH
BR     LED_OUT
SKT    LEDDIG.3      ;基本時間を表示
BR     LED40
SKT    LEDDIG.2
BR     LED5TIM
MOV    HL, #LED_0   ;LED6,7←"0"を表示
LED40:
MOV    HL, #LED_
BR     LED_OUT

LED5TIM:
MOV    X, #0H          ;LED5←時間の100の位を表示
MOV    A, T_TIM
MOV    BC, #HIGH LED_T
CALL   !TABLE
MOV    HL, XA

LED_OUT:
MOV    A, #1111B        ;ティグジット出力をオフ
OUT    PORT3, A
MOV    XA, HL

```

```
OUT    PORT4,XA      ;セグメントを出力  
MOV    A,LEDDIG  
OUT    PORT3,A      ;ディジット出力  
RET
```

```
$TT='TABLE'
;*****
;*** テーブル参照エリア ***
;*****
```

RM CSEG PAGE

RM_T:

```
;+++++
;+++ RMDATA-->KEYCODE +++
;+++++
DB 10H ;MODE
DB 11H ;UP
DB 12H ;DOWN
DB 13H ;END
DB 0FH ;WAIT
DB 0DH ;D0# +
DB 0AH ;RA# +
DB 08H ;S0#
DB 06H ;FA#
DB 03H ;RE#
DB 01H ;D0#
DB 0CH ;D0 +
DB 0BH ;SI +
DB 09H ;RA +
DB 07H ;S0
DB 05H ;FA
DB 04H ;MI
DB 02H ;RE
DB 00H ;D0
```

TONE_C CSEG PAGE

TONE_T:

```
;+++++
;+++ 音階データ ***
;+++++
DB 0F9H ;D0
DB 0EBH ;D0#
DB 0DEH ;RE
DB 0D1H ;RE#
DB 0C6H ;MI
DB 0BAH ;FA
DB 0B0H ;FA#
DB 0A6H ;S0
DB 09CH ;S0#
DB 094H ;RA +
DB 08BH ;RA# +
DB 083H ;SI +
```

```

DB      07CH          ;D0  +
DB      075H          ;D0# +
DB      0FFH          ;無音
DB      0FFH          ;無音

```

LCD_C CSEG PAGE

LCD_T:

```

;+++++LCD表示パターン+++++
;+++      LCD表示パターン      +++
;+++++LCD表示パターン+++++
DB      0EH          ;"D0  "
DB      19H
DB      2EH
DB      2EH

DB      0EH          ;;"D0#  "
DB      19H
DB      25H
DB      2EH

DB      1CH          ;"RE  "
DB      0FH
DB      2EH
DB      2EH

DB      1CH          ;;"RE#  "
DB      0FH
DB      25H
DB      2EH

DB      17H          ;"MI  "
DB      13H
DB      2EH
DB      2EH

DB      10H          ;"FA  "
DB      0BH
DB      2EH
DB      2EH

DB      10H          ;;"FA#  "
DB      0BH
DB      25H
DB      2EH

DB      1DH          ;"S0  "
DB      19H
DB      2EH
DB      2EH

```

```
DB      1DH          ;"S0#  "
DB      19H
DB      25H
DB      2EH

DB      1CH          ;"RA  +"
DB      0BH
DB      2EH
DB      26H

DB      1CH          ;"RA#  +"
DB      0BH
DB      25H
DB      26H

DB      1DH          ;"SI  +"
DB      13H
DB      2EH
DB      26H

DB      0EH          ;"DO  +"
DB      19H
DB      2EH
DB      26H

DB      0EH          ;"DO#  +"
DB      19H
DB      25H
DB      26H

DB      21H          ;"WAIT  "
DB      0BH
DB      13H
DB      1EH

DB      0FH          ;"END  "
DB      18H
DB      0EH
DB      2EH
```

LED_C CSEG PAGE

LED_T:

```
;+++++  
;+++ LEDセグメント +++  
;+++++  
;Address    abcdefg.      Display  
DB        11000101B      ;"0"  
DB        10011111B      ;"1"  
DB        00100101B      ;"2"  
DB        00001101B      ;"3"  
DB        10011001B      ;"4"  
DB        01001001B      ;"5"  
DB        01000001B      ;"6"  
DB        00011011B      ;"7"  
DB        00000001B      ;"8"  
DB        00001001B      ;"9"  
  
;+++++  
;+++ テーブル参照サブルーン +++  
;+++++  
TABLEINC:  
    INCS    DE          ;Table Address +1  
    MOV     XA,DE  
TABLE:  
    MOVT    XA,@BCXA  
    RET
```

[× ×]

アンケート記入のお願い

お手数ですが、このドキュメントに対するご意見をお寄せください。今後のドキュメント作成の参考にさせていただきます。

[ドキュメント名] μPD75008サブシリーズ アプリケーション・ノート
(U10452JJ1V0AN00 (第1版))

[お名前など] (さしつかえのない範囲で)

御社名 (学校名、その他) ()
ご住所 ()
お電話番号 ()
お仕事の内容 ()
お名前 ()

1. ご評価 (各欄に○をご記入ください)

項目	大変良い	良い	普通	悪い	大変悪い
全体の構成					
説明内容					
用語解説					
調べやすさ					
デザイン、字の大きさなど					
その他 ()					
()					

2. わかりやすい所 (第 章、第 章、第 章、第 章、その他)

理由 []

3. わかりにくい所 (第 章、第 章、第 章、第 章、その他)

理由 []

4. ご意見、ご要望

5. このドキュメントをお届けしたのは

NEC販売員、特約店販売員、NEC半導体ソリューション技術本部員、
その他 ()

ご協力ありがとうございました。

下記あてにFAXで送信いただくか、最寄りの販売員にコピーをお渡しください。

NEC半導体インフォメーションセンター
FAX: (044) 548-7900

お問い合わせは、最寄りのNECへ

【営業関係お問い合わせ先】

半導体第一販売事業部	〒108-01 東京都港区芝五丁目7番1号(NEC本社ビル)	東京 (03)3454-1111 (大代表)
半導体第三販売事業部	〒460 名古屋市中区錦一丁目17番1号(NEC中部ビル)	名古屋 (052)222-2170
中部支社 半導体販売部	〒460 名古屋市中区錦一丁目17番1号(NEC中部ビル)	名古屋 (052)222-2170
関西支社 半導体第二販売部	〒540 大阪市中央区城見一丁目4番24号(NEC関西ビル)	大阪 (06) 945-3178 大阪 (06) 945-3200 大阪 (06) 945-3208
北海道支社 札幌 (011)231-0161 東北支社 仙台 (022)261-5511 岩手支社 盛岡 (0196)51-4344 岩形支店 日高 (0236)23-5511 山形支店 恵庭 (0249)23-5511 福島支店 いわき (0246)21-5511 長野支店 立川 (0258)36-2155 滋賀支店 玉瀬 (0298)23-6161 愛知支店 水戸 (0292)26-1717 神奈川支社 横浜 (045)324-5511 静岡支店 高崎 (0273)26-1255 大分支店 太田 (0276)46-4011 宇都宮支店 宇都宮 (0286)21-2281	小山支店 小山 (0285)24-5011 長野支店 松本 (0262)35-1444 上諏訪支店 諏訪 (0263)35-1666 甲府支店 甲府 (0266)53-5350 諏訪支店 諏訪 (0552)24-4141 塩尻支店 塩尻 (048)641-1411 立川支店 立川 (0425)26-5981 千葉支店 千葉 (043)238-8116 静岡支店 静岡 (054)255-2211 浜松支店 浜松 (053)452-2711 北陸支店 金沢 (0762)23-1621 福井支店 福井 (0776)22-1866 富山支店 富山 (0764)31-8461	三重支店 津 (0592)25-7341 京都支店 京都 (075)344-7824 神戸支店 神戸 (078)333-3854 中四国支店 広島 (082)242-5504 鳥取支店 鳥取 (0857)27-5311 岡山支店 岡山 (086)225-4455 高松支店 高松 (0878)36-1200 新潟支店 新潟 (0897)32-5001 松山支店 松山 (0899)45-4111 九州支店 福岡 (092)271-7700 北九州支店 北九州 (093)541-2887

【本資料に関する技術お問い合わせ先】

半導体ソリューション技術本部 マイクロコンピュータ技術部	〒210 川崎市幸区幸越三丁目484番地	川崎 (044)548-7923
半導体販売技術本部 東日本販売技術部	〒108-01 東京都港区芝五丁目7番1号(NEC本社ビル)	東京 (03)3798-9619
半導体販売技術本部 中部販売技術部	〒460 名古屋市中区錦一丁目17番1号(NEC中部ビル)	名古屋 (052)222-2125
半導体販売技術本部 西日本販売技術部	〒540 大阪市中央区城見一丁目4番24号(NEC関西ビル)	大阪 (06) 945-3383
半導体インフォメーションセンター FAX(044)548-7900 (FAXにてお頼い致します)		